
    barrel_mcp

MCP (Model Context Protocol) library for Erlang. Implements the
MCP specification (protocol 2025-11-25 with downward negotiation
through 2024-11-05) for both server and client modes, including
the Streamable HTTP transport for Claude Code and any other MCP
client.
Features
	Full MCP Protocol: tools, resources, resource templates
(with runtime RFC 6570 expansion on resources/read),
prompts, completions, sampling, tasks (long-running
operations), _meta extension hook end-to-end, notifications
(*/list_changed, progress, cancelled,
resources/updated, tasks/status, replay_truncated).
	Tool handlers: arity 1 or arity 2 ((Args, Ctx)) with
Ctx-driven progress, cancel, and _meta hooks. Return
shapes: plain content, {tool_error, ...} (→ isError: true),
{structured, Data, ...} (→ structuredContent), or any of
the meta-bearing variants that attach _meta to the response.
	Schema validation: opt-in validate_input /
validate_output against registered JSON Schemas
(barrel_mcp_schema).
	Transports: Streamable HTTP (Claude Code), legacy HTTP,
stdio (Claude Desktop). The HTTP server is built on h1/h2
(HTTP/1.1 + HTTP/2 on one port via ALPN) — no Cowboy. Streamable
HTTP defaults to 127.0.0.1, validates Origin, and replays SSE
events via Last-Event-ID.
	Authentication: bearer (JWT/opaque), API keys (peppered
HMAC-SHA-256), basic (PBKDF2-SHA256), custom providers.
Constant-time hash comparison; legacy SHA-256 hex digests still
verify for one release. RFC 9728 Protected Resource Metadata
endpoint with spec-correct WWW-Authenticate for OAuth client
auto-discovery.
	Client library (barrel_mcp_client): supervised
gen_statem with stdio + Streamable HTTP transports, OAuth 2.1	PKCE, federation registry (one connection per server id),
pagination, schema pre-flight.


	Zero JSON dependency: uses OTP 27+ built-in json module.

Installation
Add to your rebar.config:
{deps, [
    {barrel_mcp,
        {git, "https://github.com/barrel-platform/barrel_mcp.git",
              {tag, "v2.0.2"}}}
]}.
Track main instead of pinning a tag for the latest fixes:
{barrel_mcp,
    {git, "https://github.com/barrel-platform/barrel_mcp.git",
          {branch, "main"}}}
Architecture
barrel_mcp uses a supervised gen_statem process to manage the handler registry:
	Writes (reg/unreg) go through the gen_statem for atomic operations
	Reads (find/all/run) use persistent_term directly for O(1) lookups
	States: not_ready → ready for flexible initialization
	Postpone pattern: Calls in not_ready state are postponed until ready

┌─────────────────────────────────────────────────────────────────┐
│                       barrel_mcp_sup                             │
│                      (supervisor)                                │
└─────────────────────────────────────────────────────────────────┘
                              │
                              ▼
┌─────────────────────────────────────────────────────────────────┐
│                    barrel_mcp_registry                           │
│                      (gen_statem)                                │
│                                                                  │
│  States: not_ready ──────────────────► ready                    │
│              │         (self ! ready)                            │
│              │              or                                   │
│              └──── wait for external process ────►               │
│                                                                  │
│  ┌─────────────┐        ┌─────────────────────────────────────┐ │
│  │  ETS Table  │───────►│     persistent_term (read-only)     │ │
│  │ (authority) │  sync  │         O(1) lookups                │ │
│  └─────────────┘        └─────────────────────────────────────┘ │
└─────────────────────────────────────────────────────────────────┘
         ▲                              │
         │ reg/unreg                    │ find/all/run
         │ (atomic, postponed           │ (lock-free)
         │  if not ready)               │
Configuration
To make the registry wait for an external process before becoming ready:
%% In sys.config or application env
{barrel_mcp, [
    {wait_for_proc, my_init_process}  % Wait for this process to be registered
]}.
If wait_for_proc is not set, the registry becomes ready immediately after init.
Usage by role
barrel_mcp covers the three MCP roles in one library:
	server — exposes tools, resources, prompts to MCP clients.
	client — connects to one MCP server, calls tools, reads
resources, handles server-initiated requests.
	host (agent) — drives one or more clients on behalf of an
LLM; collects each server's tool catalog, hands it to the
model, routes the model's tool call back through the right
client.

The three short examples below cover the typical wiring; deeper
guides live under guides/ (getting-started.md,
tools-resources-prompts.md, building-a-client.md).
Server — expose a tool over Streamable HTTP
-module(my_server).
-export([start/0, search/1]).

start() ->
    {ok, _} = application:ensure_all_started(barrel_mcp),
    ok = barrel_mcp:reg_tool(<<"search">>, ?MODULE, search, #{
        description => <<"Search the index">>,
        input_schema => #{<<"type">> => <<"object">>,
                           <<"required">> => [<<"q">>]}
    }),
    {ok, _} = barrel_mcp:start_http_stream(#{port => 8080,
                                              session_enabled => true}),
    ok.

search(#{<<"q">> := Q}) ->
    iolist_to_binary([<<"results for ">>, Q]).
That's a complete MCP server. Point any MCP client (Claude Code,
Claude Desktop via stdio, the barrel_mcp_client below, …) at
http://127.0.0.1:8080/mcp.
Client — connect and call a tool
client_demo() ->
    {ok, _} = application:ensure_all_started(barrel_mcp),
    {ok, Pid} = barrel_mcp_client:start(#{
        transport => {http, <<"http://127.0.0.1:8080/mcp">>}
    }),
    {ok, Result} = barrel_mcp_client:call_tool(
                     Pid, <<"search">>, #{<<"q">> => <<"hello">>}),
    barrel_mcp_client:close(Pid),
    Result.
The transport tuple selects the wire ({http, Url},
{stdio, [Cmd | Args]}). Auth and OAuth options live on the same
spec — see guides/building-a-client.md.
Host (agent) — hand many MCP servers to an LLM
agent_loop() ->
    {ok, _} = application:ensure_all_started(barrel_mcp),
    {ok, _} = barrel_mcp:start_client(<<"github">>, #{
        transport => {http, <<"https://mcp.github.example/mcp">>},
        auth => {bearer, GhToken}
    }),
    {ok, _} = barrel_mcp:start_client(<<"shell">>, #{
        transport => {stdio, ["mcp-shell-server"]}
    }),
    %% Hand every connected server's tools to the model:
    AnthropicTools = barrel_mcp_agent:to_anthropic(),
    %% ... call the LLM with AnthropicTools and capture the
    %%     tool_use block it returned ...
    Block = ask_llm(AnthropicTools),
    {NsName, Args} = barrel_mcp_tool_format:from_anthropic_call(Block),
    %% Routes "github:..." to the github client, "shell:..." to
    %% the shell client.
    barrel_mcp_agent:call_tool(NsName, Args).
barrel_mcp_agent namespaces tool names as
<<"ServerId:ToolName">> across the federation.
barrel_mcp_tool_format translates between MCP tool maps and the
provider shapes (Anthropic Messages API, OpenAI Chat Completions);
swap to_anthropic/0 and from_anthropic_call/1 for the OpenAI
counterparts to use a different model. ask_llm/1 is your own
LLM HTTP call — barrel_mcp does not bundle an LLM SDK.
Quick Start
Starting the Application
%% Start barrel_mcp application
application:ensure_all_started(barrel_mcp).

%% Wait for registry to be ready (optional, for custom initialization)
ok = barrel_mcp_registry:wait_for_ready().
Registering Tools
%% Register a tool
barrel_mcp:reg_tool(<<"search">>, my_module, search, #{
    description => <<"Search for items">>,
    input_schema => #{
        type => <<"object">>,
        properties => #{
            query => #{type => <<"string">>, description => <<"Search query">>},
            limit => #{type => <<"integer">>, default => 10}
        },
        required => [<<"query">>]
    }
}).

%% Your handler function (must accept a map and be exported with arity 1)
-module(my_module).
-export([search/1]).

search(#{<<"query">> := Query} = Args) ->
    Limit = maps:get(<<"limit">>, Args, 10),
    %% Return binary, map, or list of content blocks
    <<"Found results for: ", Query/binary>>.
Registering Resources
barrel_mcp:reg_resource(<<"config">>, my_module, get_config, #{
    name => <<"Application Config">>,
    uri => <<"config://app/settings">>,
    description => <<"Application configuration">>,
    mime_type => <<"application/json">>
}).
Registering Prompts
barrel_mcp:reg_prompt(<<"summarize">>, my_module, summarize_prompt, #{
    description => <<"Summarize content">>,
    arguments => [
        #{name => <<"content">>, description => <<"Content to summarize">>, required => true},
        #{name => <<"style">>, description => <<"Summary style">>, required => false}
    ]
}).

%% Handler returns prompt messages
summarize_prompt(Args) ->
    Content = maps:get(<<"content">>, Args),
    #{
        description => <<"Summarize the following content">>,
        messages => [
            #{role => <<"user">>, content => #{type => <<"text">>, text => Content}}
        ]
    }.
Starting Streamable HTTP Server (Claude Code)
For Claude Code integration, use the Streamable HTTP transport:
%% Start Streamable HTTP server on port 9090
{ok, _} = barrel_mcp:start_http_stream(#{port => 9090}).

%% With API key authentication
{ok, _} = barrel_mcp:start_http_stream(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{
            keys => #{<<"my-key">> => #{subject => <<"user">>}}
        }
    }
}).
Then add to Claude Code:
claude mcp add my-server --transport http http://localhost:9090/mcp \
  --header "X-API-Key: my-key"

See guides/http-stream.md for full documentation.
Starting HTTP Server (Legacy)
%% Start HTTP server on port 9090
{ok, _} = barrel_mcp:start_http(#{port => 9090}).

%% Or with custom IP binding
{ok, _} = barrel_mcp:start_http(#{port => 9090, ip => {127, 0, 0, 1}}).
Authentication
barrel_mcp provides pluggable authentication following OAuth 2.1 patterns as recommended by the MCP specification. Authentication is optional and configurable per HTTP server.
Built-in Providers
	Provider	Description
	barrel_mcp_auth_none	No authentication (default)
	barrel_mcp_auth_bearer	Bearer token (JWT or opaque)
	barrel_mcp_auth_apikey	API key authentication
	barrel_mcp_auth_basic	HTTP Basic authentication
	barrel_mcp_auth_custom	Custom auth module (simple interface)

Bearer Token (JWT) Authentication
%% Start HTTP server with JWT authentication
{ok, _} = barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_bearer,
        provider_opts => #{
            secret => <<"your-jwt-secret-key">>,
            issuer => <<"https://auth.example.com">>,
            audience => <<"https://mcp.example.com">>,
            clock_skew => 60  % seconds
        },
        required_scopes => [<<"mcp:read">>, <<"mcp:write">>]
    }
}).
For RS256/ES256 or opaque tokens, use a custom verifier:
%% Custom token verifier (e.g., for token introspection)
Verifier = fun(Token) ->
    case my_auth_service:validate(Token) of
        {ok, Claims} -> {ok, Claims};
        error -> {error, invalid_token}
    end
end,

{ok, _} = barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_bearer,
        provider_opts => #{verifier => Verifier}
    }
}).
API Key Authentication
%% Simple API key list
{ok, _} = barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{
            keys => #{
                <<"key-abc123">> => #{subject => <<"user1">>, scopes => [<<"read">>]},
                <<"key-xyz789">> => #{subject => <<"user2">>, scopes => [<<"read">>, <<"write">>]}
            }
        }
    }
}).

%% With hashed keys for security (recommended for production)
HashedKey = barrel_mcp_auth_apikey:hash_key(<<"my-secret-key">>),
{ok, _} = barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{
            keys => #{HashedKey => #{subject => <<"user1">>}},
            hash_keys => true
        }
    }
}).
Basic Authentication
%% Simple username/password
{ok, _} = barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_basic,
        provider_opts => #{
            credentials => #{
                <<"admin">> => <<"password123">>,
                <<"user">> => <<"secret">>
            },
            realm => <<"MCP Server">>
        }
    }
}).

%% With hashed passwords (recommended)
HashedPwd = barrel_mcp_auth_basic:hash_password(<<"my-password">>),
{ok, _} = barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_basic,
        provider_opts => #{
            credentials => #{<<"admin">> => HashedPwd},
            hash_passwords => true
        }
    }
}).
Custom Authentication (Simple Interface)
For integrating with existing auth systems, use barrel_mcp_auth_custom with a simple two-function module:
-module(my_auth).
-export([init/1, authenticate/2]).

init(_Opts) ->
    {ok, #{}}.

authenticate(Token, State) ->
    case my_key_store:validate(Token) of
        {ok, Info} ->
            {ok, #{subject => Info}, State};
        error ->
            {error, invalid_token, State}
    end.
Configure it:
{ok, _} = barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_custom,
        provider_opts => #{
            module => my_auth
        }
    }
}).
See guides/custom-authentication.md for full documentation.
Custom Authentication Provider (Full Behaviour)
For more control, implement the full barrel_mcp_auth behaviour:
-module(my_auth_provider).
-behaviour(barrel_mcp_auth).

-export([init/1, authenticate/2, challenge/2]).

init(Opts) ->
    {ok, Opts}.

authenticate(Request, State) ->
    Headers = maps:get(headers, Request, #{}),
    case barrel_mcp_auth:extract_bearer_token(Headers) of
        {ok, Token} ->
            %% Your validation logic
            case validate_with_my_service(Token) of
                {ok, User} ->
                    {ok, #{
                        subject => User,
                        scopes => [<<"read">>],
                        claims => #{}
                    }};
                error ->
                    {error, invalid_token}
            end;
        {error, no_token} ->
            {error, unauthorized}
    end.

challenge(Reason, _State) ->
    {401, #{<<"www-authenticate">> => <<"Bearer realm=\"mcp\"">>}, <<>>}.
Accessing Auth Info in Handlers
Authentication info is available in the request context:
my_tool_handler(Args) ->
    %% Auth info is passed in the _auth key
    case maps:get(<<"_auth">>, Args, undefined) of
        undefined ->
            <<"No auth info">>;
        AuthInfo ->
            Subject = maps:get(subject, AuthInfo),
            <<"Hello ", Subject/binary>>
    end.
Starting stdio Server (for Claude Desktop)
%% This blocks and handles MCP over stdin/stdout
barrel_mcp:start_stdio().
Using as Client
barrel_mcp_client is a gen_statem. Start it, wait for the
handshake to complete, call tools.
{ok, Pid} = barrel_mcp_client:start_link(#{
    transport => {http, <<"http://localhost:9090/mcp">>}
}),
{ok, Tools}  = barrel_mcp_client:list_tools(Pid),
{ok, Result} = barrel_mcp_client:call_tool(Pid, <<"search">>,
                                           #{<<"query">> => <<"hello">>}),
ok = barrel_mcp_client:close(Pid).
For the full task-oriented walkthrough — transport choice, auth,
OAuth, server-to-client handlers, federation, schema validation —
see Building a client. For
architecture and behaviour contracts, see
Internals. Three runnable examples live
under examples/ on
GitHub
(echo_client, sampling_host, agent_host).
Claude Desktop Configuration
When using barrel_mcp with stdio transport for Claude Desktop:
{
  "mcpServers": {
    "my-server": {
      "command": "/path/to/my_app/bin/my_app",
      "args": ["mcp"]
    }
  }
}
Your application's entry point should call barrel_mcp:start_stdio().
API Reference
Tools
	Function	Description
	barrel_mcp:reg_tool(Name, Module, Function, Opts)	Register a tool
	barrel_mcp:unreg_tool(Name)	Unregister a tool
	barrel_mcp:call_tool(Name, Args)	Call a tool locally
	barrel_mcp:list_tools()	List all registered tools

Resources
	Function	Description
	barrel_mcp:reg_resource(Name, Module, Function, Opts)	Register a resource
	barrel_mcp:unreg_resource(Name)	Unregister a resource
	barrel_mcp:read_resource(Name)	Read a resource locally
	barrel_mcp:list_resources()	List all registered resources

Prompts
	Function	Description
	barrel_mcp:reg_prompt(Name, Module, Function, Opts)	Register a prompt
	barrel_mcp:unreg_prompt(Name)	Unregister a prompt
	barrel_mcp:get_prompt(Name, Args)	Get a prompt locally
	barrel_mcp:list_prompts()	List all registered prompts

Registry
	Function	Description
	barrel_mcp_registry:start_link()	Start the registry (called by supervisor)
	barrel_mcp_registry:wait_for_ready()	Wait for registry to be ready
	barrel_mcp_registry:wait_for_ready(Timeout)	Wait with custom timeout

Server
	Function	Description
	barrel_mcp:start_http_stream(Opts)	Start Streamable HTTP server (Claude Code)
	barrel_mcp:stop_http_stream()	Stop Streamable HTTP server
	barrel_mcp:start_http(Opts)	Start HTTP server (legacy)
	barrel_mcp:stop_http()	Stop HTTP server
	barrel_mcp:start_stdio()	Start stdio server (blocking)

Client
	Function	Description
	barrel_mcp_client:connect(Opts)	Connect to MCP server
	barrel_mcp_client:initialize(Client)	Initialize connection
	barrel_mcp_client:list_tools(Client)	List available tools
	barrel_mcp_client:call_tool(Client, Name, Args)	Call a tool
	barrel_mcp_client:list_resources(Client)	List available resources
	barrel_mcp_client:read_resource(Client, Uri)	Read a resource
	barrel_mcp_client:list_prompts(Client)	List available prompts
	barrel_mcp_client:get_prompt(Client, Name, Args)	Get a prompt
	barrel_mcp_client:close(Client)	Close connection

Authentication
	Function	Description
	barrel_mcp_auth:extract_bearer_token(Headers)	Extract Bearer token from headers
	barrel_mcp_auth:extract_api_key(Headers, Opts)	Extract API key from headers
	barrel_mcp_auth:extract_basic_auth(Headers)	Extract Basic auth credentials
	barrel_mcp_auth_apikey:hash_key(Key)	Hash an API key (SHA256)
	barrel_mcp_auth_basic:hash_password(Password)	Hash a password (SHA256)

MCP Protocol Support
Supported Methods
Lifecycle:
	initialize / initialized
	ping

Tools:
	tools/list
	tools/call

Resources:
	resources/list
	resources/read
	resources/templates/list
	resources/subscribe / resources/unsubscribe

Prompts:
	prompts/list
	prompts/get

Sampling:
	sampling/createMessage

Logging:
	logging/setLevel

Pending features
Design notes and scoped sketches for things not yet implemented
live in docs/pending-features.md.
First entry: Enterprise-Managed Authorization (the second
half of the MCP ext-auth extension — RFC 8693 token-exchange
chained through an org IdP for SSO-driven MCP access). Open an
issue if you want one prioritised.
Development
# Compile
rebar3 compile

# Run tests
rebar3 eunit

# Dialyzer
rebar3 dialyzer

# Shell
rebar3 shell

License
Apache-2.0


  

    Getting Started

This guide will help you get started with barrel_mcp, an Erlang library implementing
the Model Context Protocol (MCP) specification.
Installation
Add barrel_mcp to your rebar.config dependencies:
{deps, [
    {barrel_mcp, "1.0.0"}
]}.
Or from git:
{deps, [
    {barrel_mcp, {git, "https://github.com/barrel-db/barrel_mcp.git", {tag, "v1.0.0"}}}
]}.
Starting the Application
%% Start the barrel_mcp application
application:ensure_all_started(barrel_mcp).

%% Wait for registry to be ready (recommended)
ok = barrel_mcp_registry:wait_for_ready().
Your First MCP Server
1. Register a Tool
Tools are functions that can be called by MCP clients (like Claude):
-module(my_tools).
-export([greet/1]).

%% Arity-1 handler — gets only the call arguments. The simplest
%% shape; pick this when you don't need progress, cancellation, or
%% session context. For arity-2 handlers (with a Ctx parameter),
%% see the Tools guide.
greet(Args) ->
    Name = maps:get(<<"name">>, Args, <<"World">>),
    <<"Hello, ", Name/binary, "!">>.
Register it:
barrel_mcp:reg_tool(<<"greet">>, my_tools, greet, #{
    description => <<"Greet someone by name">>,
    input_schema => #{
        <<"type">> => <<"object">>,
        <<"properties">> => #{
            <<"name">> => #{
                <<"type">> => <<"string">>,
                <<"description">> => <<"Name to greet">>
            }
        }
    }
}).
2. Start the HTTP Server
{ok, _} = barrel_mcp:start_http_stream(#{port => 9090}).
The Streamable HTTP server binds to 127.0.0.1 by default and
validates the Origin header on every request. Public binds (any
non-loopback IP) require an explicit allowed_origins — see the
Streamable HTTP guide for the security defaults.
The legacy plain-HTTP transport (start_http/1, JSON-RPC only,
protocol 2024-11-05) is still available for clients that don't
speak the Streamable HTTP transport.
3. Test Your Server
Using curl:
# Initialize
curl -X POST http://localhost:9090/mcp \
  -H "Content-Type: application/json" \
  -d '{"jsonrpc":"2.0","id":1,"method":"initialize","params":{}}'

# List tools
curl -X POST http://localhost:9090/mcp \
  -H "Content-Type: application/json" \
  -d '{"jsonrpc":"2.0","id":2,"method":"tools/list"}'

# Call your tool
curl -X POST http://localhost:9090/mcp \
  -H "Content-Type: application/json" \
  -d '{"jsonrpc":"2.0","id":3,"method":"tools/call","params":{"name":"greet","arguments":{"name":"Erlang"}}}'

Using with Claude Desktop
For Claude Desktop integration, use the stdio transport:
	Create an escript or release that calls barrel_mcp:start_stdio()
	Configure Claude Desktop's claude_desktop_config.json:

{
  "mcpServers": {
    "my-server": {
      "command": "/path/to/my_app",
      "args": ["mcp"]
    }
  }
}
Next Steps
	Tools, Resources & Prompts — handler shapes (arity 1 / arity 2 with Ctx), tool errors, structured output, long-running tasks, completions, resource templates, server notifications.
	Streamable HTTP transport — security defaults, CORS, Origin validation, response codes, replay.
	Authentication — bearer / API key / basic, modern hash formats.
	Features matrix — what's supported on the wire.
	Building a client — task-oriented walkthrough for hosting MCP clients.
	Client internals — architecture and behaviour contracts.
	MCP Client (reference) - Older API reference, kept for cross-linking



  

    Building an MCP Client with barrel_mcp

barrel_mcp is a pure MCP library. It implements the wire protocol,
the transports, and the client state machine. It does not call
LLM providers, build prompts, or run an agent loop — those belong to
the host application that uses this library.
This guide is task-oriented. Each section answers "I want to do X"
with a working snippet, notes, and a pointer to the spec or wire
detail. Snippets tagged ```erlang are extracted from this file
and compile-checked in CI; snippets tagged ```erl are
illustrative output only.
If you have not yet read the architecture, the short
summary is: a barrel_mcp_client is a gen_statem that owns one
connection to one MCP server. It dispatches inbound responses to
waiting callers, server-initiated requests to a host-supplied handler
module, and notifications to either subscribers or the same handler.

1. What barrel_mcp gives you, and what it doesn't
It gives you:
	Streamable HTTP and stdio transports.
	A spec-conformant MCP client (barrel_mcp_client).
	Server-to-client request dispatch via the
barrel_mcp_client_handler
behaviour.
	OAuth 2.1 + PKCE primitives (RFC 9728, RFC 8414, RFC 8707) and a
refresh-only auth handle.
	A federation registry (barrel_mcp_clients) for hosting many MCP
connections in one app.
	A JSON Schema subset validator (barrel_mcp_schema).

It does not give you:
	LLM provider HTTP (Anthropic, OpenAI, Hermes, etc.). Implement that
in your host application.
	An agent loop. Drive your own multi-turn loop using the building
blocks below.
	Tool-name namespacing across servers. That's host policy.
	A browser-based redirect listener for OAuth. Hosts run that step
with whatever UI fits their environment.


2. Choose a transport
	Transport	Use it when	Where it lives
	Streamable HTTP	The server is remote or runs as a long-lived service. You want session resumption and server-initiated requests over SSE.	barrel_mcp_client_http
	stdio	The server is a local subprocess (CLI tools, native MCP servers shipped as binaries).	barrel_mcp_client_stdio

In both cases the high-level API on barrel_mcp_client is identical.
The transport tuple in the connect spec is the only difference:
%% Streamable HTTP
#{transport => {http, <<"https://server.example/mcp">>}}

%% stdio
#{transport => {stdio, #{command => "/usr/local/bin/mcp-server",
                         args => ["--quiet"]}}}

3. Connect spec reference
barrel_mcp_client:start_link/1 and start/1 accept a single map.
Every key is documented below.
	Key	Type	Default	Effect
	transport	{http, Url} | {stdio, #{command, args}}	required	Which transport to open.
	client_info	#{name, version}	#{name => <<"barrel_mcp_client">>, version => <<"2.0.2">>}	Sent in initialize.
	capabilities	map	#{}	Client capabilities to declare. Booleans become spec-shape objects on the wire (e.g. #{sampling => true} becomes #{<<"sampling">> => #{}}).
	handler	{Mod, Args}	{barrel_mcp_client_handler_default, []}	Module implementing barrel_mcp_client_handler to handle server-initiated requests and notifications.
	auth	none | {bearer, Token} | {oauth, Config}	none	Authentication. See section 14.
	protocol_version	binary	?MCP_CLIENT_PROTOCOL_VERSION (<<"2025-11-25">>)	Target protocol version. The client negotiates downward if the server reports an older one.
	request_timeout	pos_integer	30000	Default per-request timeout in ms.
	init_timeout	pos_integer	30000	Time allowed for the initialize round-trip.
	ping_interval	pos_integer | infinity	infinity	If set, the client sends ping every N ms while in ready.
	ping_failure_threshold	pos_integer	3	Consecutive ping failures before the connection is closed with reason ping_failed.


4. Connect and close
{ok, Client} = barrel_mcp_client:start_link(#{
    transport => {http, <<"http://127.0.0.1:9090/mcp">>}
}),
%% ... use Client ...
ok = barrel_mcp_client:close(Client).
start_link/1 links the calling process to the client. Use start/1
for unsupervised one-offs (tests, scripts).
The state machine moves connecting → initializing → ready. Calls
made before ready return {error, not_ready}. Wait for the first
server_capabilities/1 to succeed if you need to gate work on
readiness:
wait_ready(Pid, 0) -> {error, not_ready};
wait_ready(Pid, N) ->
    case catch barrel_mcp_client:server_capabilities(Pid) of
        {ok, _} -> ok;
        _ ->
            timer:sleep(100),
            wait_ready(Pid, N - 1)
    end.

5. Capability negotiation and version downgrade
The client declares what it can answer (sampling, roots, elicitation)
and the server replies with what it can serve (tools, resources,
prompts, logging, completions). After the handshake:
get_caps(Pid) ->
    {ok, ServerCaps} = barrel_mcp_client:server_capabilities(Pid),
    case maps:is_key(<<"tools">>, ServerCaps) of
        true -> ok;
        false -> {error, no_tools}
    end.
Version is negotiated automatically. The client sends
?MCP_CLIENT_PROTOCOL_VERSION (<<"2025-11-25">> today); if the
server replies with an older version (e.g. 2025-03-26), the client
adopts that and uses it on every subsequent MCP-Protocol-Version
header. Read it back with barrel_mcp_client:protocol_version/1.

6. List tools, resources, and prompts
Single-page calls return one chunk:
list_one_page(Pid) ->
    {ok, Tools} = barrel_mcp_client:list_tools(Pid),
    Tools.
Pagination is opt-in. Pass #{want_cursor => true} to receive
{ok, Items, NextCursor | undefined}. If the server has more pages
than you want to walk by hand, use the *_all helpers:
list_every_tool(Pid) ->
    {ok, AllTools} = barrel_mcp_client:list_tools_all(Pid),
    AllTools.
The same shape applies to list_resources/1,2,
list_resource_templates/1,2, and list_prompts/1,2.
Translate to provider tool shapes
barrel_mcp_tool_format converts MCP tool maps to the shapes the
LLM provider APIs expect, and vice versa. Use it when you're
building an agent host that hands MCP tools to Anthropic / OpenAI
models and routes the model's tool calls back through the MCP
client.
single_server_loop(McpPid) ->
    {ok, McpTools} = barrel_mcp_client:list_tools_all(McpPid),
    AnthropicTools = barrel_mcp_tool_format:to_anthropic(McpTools),
    %% ... call Anthropic with AnthropicTools, get tool_use blocks ...
    Block = receive_tool_use_block(),
    {Name, Args} = barrel_mcp_tool_format:from_anthropic_call(Block),
    barrel_mcp_client:call_tool(McpPid, Name, Args).
receive_tool_use_block() ->
    %% Replace with your real LLM client.
    #{<<"name">> => <<"echo">>,
      <<"input">> => #{<<"text">> => <<"hi">>}}.
to_openai/1 and from_openai_call/1 follow the same pattern for
the OpenAI Chat Completions tool-call envelope.
Federate across servers with the agent aggregator
When the host connects to several MCP servers at once (via
barrel_mcp:start_client/2), barrel_mcp_agent collapses every
catalog into one namespaced list and routes a model's call back to
the right server.
multi_server_loop() ->
    Tools = barrel_mcp_agent:to_anthropic(),
    %% ... call Anthropic with Tools ...
    Block = receive_tool_use_block(),
    {Name, Args} = barrel_mcp_tool_format:from_anthropic_call(Block),
    barrel_mcp_agent:call_tool(Name, Args).
Tool names round-trip through <<"ServerId:ToolName">>. Pick a
different separator with #{separator => <<"::">>} if : clashes
with one of your tool names.

7. Call a tool
call_echo(Pid) ->
    barrel_mcp_client:call_tool(Pid, <<"echo">>, #{<<"text">> => <<"hi">>}).
call_tool/4 accepts an option map:
call_with_progress(Pid, Token) ->
    barrel_mcp_client:call_tool(
        Pid,
        <<"slow">>,
        #{<<"size">> => 1000},
        #{progress_token => Token, timeout => 60000}).
When progress_token is supplied, the calling process receives one
{mcp_progress, Token, Params} message per notifications/progress
the server emits, until the request settles (response, cancel, or
timeout).
The full echo-client example lives in
examples/echo_client/src/echo_client.erl.
Tool results
call_tool/3,4 returns the server's result map. Three shapes
the spec allows:
classify(#{<<"isError">> := true} = R) ->
    {error, maps:get(<<"content">>, R)};
classify(#{<<"structuredContent">> := Data} = R) ->
    {structured, Data, maps:get(<<"content">>, R, [])};
classify(#{<<"content">> := Content}) ->
    {ok, Content}.
	isError: true → the tool reported a domain-level failure
(validation, business rule). The content is human-readable.
	structuredContent → typed payload, optionally paired with
human-readable content blocks. When the tool registered an
outputSchema, the typed payload conforms to it.
	Plain content → standard MCP content blocks.

Tasks
If the server registered the tool with long_running => true,
call_tool returns immediately with #{<<"taskId">> := Id, <<"status">> := <<"working">>}. Track progress with the methods
in section 12.

8. Read and subscribe to resources
read_resource(Pid, Uri) ->
    barrel_mcp_client:read_resource(Pid, Uri).
Subscribe to be notified of updates:
watch_resource(Pid, Uri) ->
    {ok, _} = barrel_mcp_client:subscribe(Pid, Uri),
    receive
        {mcp_resource_updated, Uri, Params} ->
            handle_update(Params)
    after 5000 ->
        timeout
    end.
handle_update(_) -> ok.
The subscription stays in the client's state until you call
unsubscribe(Pid, Uri) or close the client. Subscribers are
identified by their pid; multiple processes can subscribe to the same
URI on the same client.

Logging
Set the server's log level for the session, and route the
inbound notifications/message stream into your application's
logger via the handler.
set_debug_level(Pid) ->
    barrel_mcp_client:set_log_level(Pid, <<"debug">>).
Levels match RFC 5424 names: debug, info, notice,
warning, error, critical, alert, emergency.
Server introspection
caps(Pid) ->
    barrel_mcp_client:server_capabilities(Pid).

info(Pid) ->
    barrel_mcp_client:server_info(Pid).

negotiated_version(Pid) ->
    barrel_mcp_client:protocol_version(Pid).
server_capabilities/1 is the authoritative source for what the
server actually supports (e.g. whether tasks is advertised).
Check before calling capability-gated methods.

9. Get prompts and run completion
fetch_prompt(Pid) ->
    barrel_mcp_client:get_prompt(Pid, <<"summarize">>,
                                 #{<<"length">> => <<"short">>}).
ask_completion(Pid) ->
    barrel_mcp_client:complete(Pid,
        #{<<"type">> => <<"ref/prompt">>, <<"name">> => <<"summarize">>},
        #{<<"name">> => <<"length">>, <<"value">> => <<"sho">>}).
complete/3 is the spec-named completion/complete request used to
auto-complete prompt argument values.

10. Handle server-initiated requests
The server can call into the client (sampling/createMessage,
roots/list, elicitation/create). Implement
barrel_mcp_client_handler and supply it as handler => {Mod, Args}.
If the host's roots change after initialize (the user opened a
new workspace, granted access to a new directory, etc.) inform
the server so it can re-query:
ok = barrel_mcp_client:notify_roots_list_changed(Pid).
The server may follow up with roots/list against your handler.
Three return shapes from handle_request/3:
	{reply, Result, State} — synchronous answer.
	{error, Code, Message, State} — JSON-RPC error response.
	{async, Tag, State} — defer; reply later from any process via
barrel_mcp_client:reply_async(Pid, Tag, Result).

Skeleton:
-module(my_handler).
-behaviour(barrel_mcp_client_handler).
-export([init/1, handle_request/3, handle_notification/3, terminate/2]).

init(Args) ->
    {ok, Args}.

handle_request(<<"sampling/createMessage">>, Params, State) ->
    Result = sample_via_llm(Params, State),
    {reply, Result, State};
handle_request(<<"roots/list">>, _, State) ->
    {reply, #{<<"roots">> => [
        #{<<"uri">> => <<"file:///workspace">>,
          <<"name">> => <<"workspace">>}
    ]}, State};
handle_request(Method, _Params, State) ->
    {error, -32601, <<"Method not found: ", Method/binary>>, State}.

handle_notification(_Method, _Params, State) ->
    {ok, State}.

terminate(_Reason, _State) ->
    ok.

sample_via_llm(_, _) ->
    %% Replace with an HTTP call to your LLM provider.
    #{<<"content">> => #{<<"type">> => <<"text">>, <<"text">> => <<"hi">>},
      <<"model">> => <<"placeholder">>,
      <<"role">> => <<"assistant">>}.
The sampling_host example in
examples/sampling_host/src/sampling_host.erl
shows the full server-to-client round-trip end to end.

11. Asynchronous handler replies
When answering a server request takes time (calling an LLM provider,
asking a user, etc.), block the model thread instead of the state
machine:
handle_request(<<"sampling/createMessage">>, Params, State) ->
    Tag = make_ref(),
    Self = self(),  %% the host process; not the gen_statem
    spawn(fun() ->
        Result = slow_llm_call(Params),
        barrel_mcp_client:reply_async(Self, Tag, Result)
    end),
    {async, Tag, State}.
reply_async/3 may also be used for errors via
reply_async(Pid, Tag, {error, Code, Message}).

12. Notifications and tasks
The handler's handle_notification/3 callback receives every inbound
notification with its raw params map. Common methods:
	notifications/resources/updated — also dispatched to subscribers
of the URI as {mcp_resource_updated, Uri, Params} (see section 8).
	notifications/progress — also dispatched to the caller of the
request that owns the progress token (see section 7).
	notifications/tools/list_changed, .../resources/list_changed,
.../prompts/list_changed — catalogue updated; re-fetch or
invalidate caches.
	notifications/tasks/status — a long-running task transitioned
state. The full task record is in params.
	notifications/message — server logging stream.
	notifications/replay_truncated — your Last-Event-ID was
outside the server's replay window; resync rather than trust the
partial stream.

The handler is the right place to integrate with your application's
metrics, logs, or UI.
Task methods
When a tool was registered as long_running on the server, the
initial call_tool returns a taskId. Track it with the typed
wrappers:
poll_task(Pid, TaskId) ->
    barrel_mcp_client:tasks_get(Pid, TaskId).

list_tasks(Pid) ->
    %% Single page; use `tasks_list_all/1' or
    %% `tasks_list/2' with `#{want_cursor => true}' for paging.
    barrel_mcp_client:tasks_list(Pid).

abort_task(Pid, TaskId) ->
    barrel_mcp_client:tasks_cancel(Pid, TaskId).

fetch_task_result(Pid, TaskId) ->
    %% Returns the recorded result for a `completed' task, or an
    %% error for `failed' / `cancelled' / still-`working' tasks.
    barrel_mcp_client:tasks_result(Pid, TaskId).
Status values on the wire are working, completed, failed,
and cancelled; createdAt and lastUpdatedAt are RFC 3339 strings.
When you registered a progress_token on the originating call,
the same task usually emits notifications/progress updates that
arrive through your handler, so polling is rarely required. Prefer
subscribing to notifications/tasks/status in the handler over
busy-polling tasks_get/2, then fetch the payload once with
tasks_result/2 when the status reaches completed.

13. Cancel, time out, ping
cancel_request(Pid, Id) ->
    barrel_mcp_client:cancel(Pid, Id).
The id is the JSON-RPC request id for the in-flight call.
barrel_mcp_client increments these internally; in tests you can
read pending ids via sys:get_state/1. In production you usually
don't cancel by id — you set a timeout on call_tool/4 and let
the deadline fire.
Periodic ping is opt-in:
%% Spec snippet — a key on barrel_mcp_client:start_link/1's input map.
#{ping_interval => 30000, ping_failure_threshold => 3}
After three consecutive ping failures (default), the connection
closes with reason ping_failed and the linked owner sees the exit.

14. Authenticate
Static bearer
#{transport => {http, <<"https://server.example/mcp">>},
  auth => {bearer, <<"my-static-token">>}}
barrel_mcp_client_auth_bearer attaches Authorization: Bearer ...
on every request. A 401 returns {error, unauthorized} and the
caller must restart with a new token.
OAuth 2.1 + PKCE
The interactive authorization-code redirect is a host concern; once
you have an access token (and ideally a refresh token), pass them
through:
#{transport => {http, <<"https://server.example/mcp">>},
  auth => {oauth, #{
    access_token   => <<"eyJ...">>,
    refresh_token  => <<"opaque">>,
    token_endpoint => <<"https://auth.example/token">>,
    client_id      => <<"my-client">>,
    resource       => <<"https://server.example/mcp">>
  }}}
On 401 the library posts a refresh_token grant to token_endpoint
(with the RFC 8707 resource parameter), updates the handle, and
retries the original request once.
To drive the initial auth code flow yourself, use the discovery
helpers:
discover(Server) ->
    {ok, Resp401, Headers} = first_request_returns_401(Server),
    Www = proplists:get_value(<<"www-authenticate">>, Headers),
    PrmUrl = barrel_mcp_client_auth_oauth:parse_www_authenticate(Www),
    {ok, Prm} = barrel_mcp_client_auth_oauth:discover_protected_resource(PrmUrl),
    [Issuer | _] = maps:get(<<"authorization_servers">>, Prm),
    {ok, AS} = barrel_mcp_client_auth_oauth:discover_authorization_server(Issuer),
    {Url, Verifier, _State} = barrel_mcp_client_auth_oauth:build_authorization_url(
        maps:get(<<"authorization_endpoint">>, AS),
        #{client_id => <<"my-client">>,
          redirect_uri => <<"http://localhost:38080/cb">>,
          resource => maps:get(<<"resource">>, Prm)}),
    {Url, Verifier, AS, Resp401}.
first_request_returns_401(_) ->
    {ok, ignore, [{<<"www-authenticate">>,
                   <<"Bearer resource_metadata=\"https://srv/.well-known/oauth-protected-resource\"">>}]}.
After the user authorizes and you capture the code from the
redirect, exchange it:
{ok, Tokens} = barrel_mcp_client_auth_oauth:exchange_code(
    maps:get(<<"token_endpoint">>, AS),
    #{code => Code,
      code_verifier => Verifier,
      client_id => <<"my-client">>,
      redirect_uri => <<"http://localhost:38080/cb">>,
      resource => maps:get(<<"resource">>, Prm)}).
Then start the client with the tokens above.

15. Schema-validate before calling
barrel_mcp_schema:validate/2 covers the JSON Schema subset MCP
tools actually use. Cache the schema returned by tools/list, then
validate before dispatching:
call_validated(Pid, Name, Args, Schema) ->
    case barrel_mcp_schema:validate(Args, Schema) of
        ok -> barrel_mcp_client:call_tool(Pid, Name, Args);
        {error, Errors} -> {error, {invalid_args, Errors}}
    end.
This is opt-in — many hosts trust the LLM output enough to skip it.
Use it when you want a clear error before the request reaches the
server.

16. Federate many MCP servers
{ok, _} = barrel_mcp:start_client(<<"github">>, #{
    transport => {http, <<"https://mcp.github.example/">>},
    auth => {bearer, GhToken}
}),
{ok, _} = barrel_mcp:start_client(<<"local-files">>, #{
    transport => {stdio, #{command => "/usr/local/bin/mcp-files"}}
}),

GitHub = barrel_mcp:whereis_client(<<"github">>),
{ok, Tools} = barrel_mcp_client:list_tools(GitHub).
Each connection is a supervised worker. Crashes are isolated; the
registry's monitor prunes the dead entry automatically. Tool-name
namespacing across servers is your call; a common pattern is
<<ServerId/binary, "::", ToolName/binary>> when surfacing the
catalogue to an LLM.

17. Errors you can see
	Return	Cause
	{error, not_ready}	Call made before the initialize handshake completed. Wait for ready.
	{error, {unsupported, Method}}	Server didn't advertise the capability the call requires.
	{error, {Code, Message}}	Server returned a JSON-RPC error.
	{error, cancelled}	Caller invoked cancel/2.
	{error, timeout}	The per-request timeout fired.
	{error, unauthorized}	401 with no usable refresh path.
	{error, {protocol_version, Server, Supported}}	Server's version is outside the client's supported list. Init failed.


18. Production checklist
	Run clients under a supervisor. barrel_mcp:start_client/2 does
this for you; for ad-hoc clients call barrel_mcp_client:start_link/1
inside your own supervision tree.
	Set request_timeout to a value that matches your SLO. The default
30 s is generous.
	Set ping_interval if your transport is a long-lived HTTP
connection that may sit idle behind proxies.
	Implement handle_notification/3 to forward notifications/message
to your logging system; this is how MCP servers emit operational
signals.
	Validate tool inputs with barrel_mcp_schema:validate/2 before
forwarding model output.
	For OAuth, persist refresh tokens; the in-memory handle dies with
the client.


See also
	Internals — architecture and behaviour contracts.
	examples/echo_client/ — minimal
end-to-end host.
	examples/sampling_host/ — handler
behaviour worked example.
	Features — spec coverage matrix.



  

    Internals

How barrel_mcp is wired on the client side. Read this if you are
extending the library, debugging a stuck connection, or generating
code that introspects the architecture.
This file is paired with Building a client;
the building guide is task-oriented, this one is structural.
1. Module map
	Module	Role
	barrel_mcp	Top-level façade. start_client/2, notify_resource_updated/1,2, sampling_create_message/3, etc.
	barrel_mcp_client	The client gen_statem. Owns one connection.
	barrel_mcp_client_sup	Supervises client workers (transient).
	barrel_mcp_clients	Federation registry: ServerId → pid().
	barrel_mcp_client_transport	Behaviour: connect/2, send/2, close/1.
	barrel_mcp_client_stdio	Transport impl over open_port/2.
	barrel_mcp_client_http	Transport impl over Streamable HTTP (POST + SSE GET).
	barrel_mcp_client_handler	Behaviour for server-initiated requests and notifications.
	barrel_mcp_client_handler_default	No-op default handler.
	barrel_mcp_client_auth	Behaviour: init/1, header/1, refresh/2.
	barrel_mcp_client_auth_bearer	Static-token impl.
	barrel_mcp_client_auth_oauth	OAuth 2.1 + PKCE impl + discovery helpers.
	barrel_mcp_protocol	JSON-RPC envelope codec; shared with the server.
	barrel_mcp_pagination	Cursor walker for */list requests.
	barrel_mcp_schema	JSON Schema subset validator.

2. Supervision tree
barrel_mcp_sup (one_for_one)
├── barrel_mcp_registry      -- server-side registry of tools/resources/prompts
├── barrel_mcp_session       -- server-side session manager
├── barrel_mcp_client_sup    -- one_for_one of barrel_mcp_client workers
│   ├── client(<<"server-1">>)
│   └── client(<<"server-2">>)
└── barrel_mcp_clients       -- registry: ServerId -> pid + monitor
barrel_mcp_clients is a gen_server that owns the lookup table
and serializes registration so two callers can't race on the same
ServerId. Lookups (whereis_client/1, list_clients/0) hit the
ETS table directly without crossing the process boundary.
3. Client state machine
barrel_mcp_client is a gen_statem in state_functions mode.
                         start_link/1
                              │
                              ▼
                       ┌────────────┐
                       │ connecting │  open transport, send `initialize'
                       └─────┬──────┘
                             │ transport up
                             ▼
                       ┌──────────────┐
                       │ initializing │  state-timeout = init_timeout
                       └─────┬────────┘
                             │ initialize response (negotiated version)
                             │ + notifications/initialized
                             │ + open SSE GET (HTTP only)
                             ▼
                       ┌────────┐
                       │ ready  │◀──── all client API calls go here
                       └─┬──┬───┘
                         │  │
              close/1 ───┘  └─── transport_closed | ping_failed
                              ▼
                       ┌─────────┐
                       │ closing │
                       └─────────┘
State transitions:
	connecting → initializing — internal event after open_transport/1.
	initializing → ready — successful initialize response, version in
?MCP_CLIENT_SUPPORTED_VERSIONS.
	initializing → stop {init_failed, _} — server returned a JSON-RPC
error to initialize or omitted protocolVersion.
	ready → closing — caller cast close, or stop on mcp_closed,
or stop with ping_failed.

4. Inbound message flow
transport process (stdio port owner | http gen_server)
        │
        │  {mcp_in, TransportPid, Json}
        ▼
barrel_mcp_client gen_statem (info handler)
        │
        ├── decode_envelope/1 -> request | response | notification | error
        │
        ├── response/error  -> ETS pending lookup -> gen_statem:reply(Caller, _)
        ├── request         -> handler:handle_request/3 -> send response
        └── notification    -> handler:handle_notification/3
                               (+ subscriber routing for resources/updated)
                               (+ progress routing for notifications/progress)
The transport process owns the wire — socket, port, SSE buffer — and
forwards one JSON-RPC envelope per {mcp_in, _, _} message. The
gen_statem never reads the wire directly. This means transports can
implement framing however suits them (line-delimited for stdio,
SSE-event-delimited for HTTP) without leaking that into the state
machine.
When the transport ends, it sends {mcp_closed, TransportPid, Reason}
and the gen_statem stops with that reason.
5. Outbound flow
caller                              barrel_mcp_client gen_statem
   │                                          │
   │ gen_statem:call({request, M, P, T})      │
   ├─────────────────────────────────────────►│
   │                                          ├─ next_id/1
   │                                          ├─ pending#{Id => #pending{...}}
   │                                          ├─ encode_request/3
   │                                          └─ transport:send/2
   │                                          ▼
   │                                   transport process
   │                                          │
   │                                          ▼
   │                                   ... server ...
   │                                          │
   │                                          ▼
   │                                   {mcp_in, _, ResponseJson}
   │                                          │
   │                                  match by id, gen_statem:reply
   │◀─────────────────────────────────────────│
pending is a map keyed by request id. Each entry tracks the caller
{From}, the method name, the deadline, and the optional progress
token. The state machine drops the entry on settle and clears any
matching state-timeout.
6. Behaviour contracts
barrel_mcp_client_transport
-callback connect(Owner :: pid(), Opts :: map()) ->
    {ok, pid()} | {error, term()}.
-callback send(TransportPid :: pid(), JsonBinary :: iodata()) ->
    ok | {error, term()}.
-callback close(TransportPid :: pid()) -> ok.
The transport process MUST emit {mcp_in, TransportPid, JsonBinary}
to Owner for every complete inbound JSON-RPC envelope. On
shutdown — peer disconnect, port exit, fatal error — it MUST emit
{mcp_closed, TransportPid, Reason} exactly once.
barrel_mcp_client_handler
-callback init(Args :: term()) -> {ok, state()} | {error, term()}.
-callback handle_request(Method :: binary(),
                         Params :: map(),
                         State :: state()) ->
    {reply, Result :: term(), state()} |
    {error, Code :: integer(), Message :: binary(), state()} |
    {async, Tag :: term(), state()}.
-callback handle_notification(Method :: binary(),
                              Params :: map(),
                              State :: state()) -> {ok, state()}.
-callback terminate(Reason :: term(), State :: state()) -> any().
The default handler (barrel_mcp_client_handler_default) returns
method_not_found for every request and ignores every notification.
Hosts only implement what they declare in their capabilities.
The {async, Tag, State} reply form lets a handler defer the
response while the state machine continues to process other inbound
traffic. The host posts the actual reply via
barrel_mcp_client:reply_async/3.
barrel_mcp_client_auth
-callback init(Config :: term()) -> {ok, handle()} | {error, term()}.
-callback header(handle()) -> {ok, binary()} | none | {error, term()}.
-callback refresh(handle(), WwwAuthenticate :: binary() | undefined) ->
    {ok, handle()} | {error, term()}.
The HTTP transport calls header/1 on every outgoing request. On a
401 it calls refresh/2 once with the WWW-Authenticate header
value, then retries the original request with the new handle. The
caller-facing constructor is barrel_mcp_client_auth:new/1 which
wraps the chosen impl as {Module, State} (or none for no auth).
7. State record fields
	Field	Type	Holds
	spec	map	The connect spec passed to start_link/1.
	transport	{Mod, Pid}	Active transport.
	request_id	int	Monotonic counter for outbound JSON-RPC ids.
	pending	map	Id ⇒ #pending{caller, method, deadline, progress_token}.
	handler_mod	atom	The host's handler module.
	handler_state	term	The handler's per-instance state.
	async_replies	map	Tag ⇒ Id for {async, Tag, _} requests.
	subscriptions	map	Uri ⇒ [pid()] for notifications/resources/updated.
	progress	map	Token ⇒ pid() for notifications/progress.
	ping_failures	int	Consecutive ping errors; resets on success.
	server_capabilities	map	What the server advertised at init.
	server_info	map	name, version from the server.
	protocol_version	binary	Negotiated version, after init.

8. Wire format reference
Streamable HTTP
Headers attached on every outgoing request:
	Header	When	Notes
	content-type: application/json	always	request body is a JSON-RPC envelope (or empty for GET).
	accept: application/json, text/event-stream	always	the server may answer with either.
	mcp-session-id: <id>	after the initialize POST returns one	echoed on every subsequent POST/GET/DELETE.
	mcp-protocol-version: <version>	after init completes	the negotiated version.
	authorization: Bearer ...	when an auth handle attaches one	bearer or OAuth-fronted.
	last-event-id: <id>	reconnecting the GET SSE	not yet a full replay path; tracked but not yet replayed.

The POST endpoint may answer with a JSON envelope or with an SSE
stream. The transport classifies on content-type and either
forwards the single envelope or parses SSE events until the matching
done. SSE event format:
id: <id>
event: <name>
data: <JSON-RPC envelope>
\n
The transport ignores event:, captures id: for resumability, and
forwards each data: payload as one mcp_in message.
The GET endpoint opens a long-lived SSE for unsolicited
server-to-client traffic. A 405 here means the server doesn't support
unsolicited streams; the client silently drops the GET and only
receives server-initiated requests interleaved on POST responses.
DELETE on close, with Mcp-Session-Id.
stdio
Line-delimited JSON-RPC. One envelope per line. The transport reads
up to a 1 MiB line limit (configurable in
barrel_mcp_client_stdio). Anything larger fails framing with
{mcp_closed, _, line_too_long}.
stdin and stdout are the only channels. stderr from the subprocess is
discarded by default; redirect it in your launcher if you need it.
9. Where to look in the source
	Want to read	File
	State machine + public API	src/barrel_mcp_client.erl
	Streamable HTTP transport	src/barrel_mcp_client_http.erl
	stdio transport	src/barrel_mcp_client_stdio.erl
	Handler behaviour + default	src/barrel_mcp_client_handler.erl, src/barrel_mcp_client_handler_default.erl
	OAuth flow	src/barrel_mcp_client_auth_oauth.erl
	Federation	src/barrel_mcp_clients.erl, src/barrel_mcp_client_sup.erl
	Pagination walker	src/barrel_mcp_pagination.erl
	Schema validator	src/barrel_mcp_schema.erl
	Wire envelope codec	src/barrel_mcp_protocol.erl



  

    barrel_mcp features

Tracks notable capabilities and the spec-conformance status of the
Erlang MCP library. See CHANGELOG.md for release-by-release detail.
Server
Transports
	HTTP transport (barrel_mcp_http) — JSON-RPC over POST. Legacy.
	Streamable HTTP transport (barrel_mcp_http_stream) — MCP
2025-11-25 with downward negotiation to 2025-06-18,
2025-03-26, 2024-11-05. POST (JSON or SSE), GET (SSE),
DELETE, OPTIONS. Default bind 127.0.0.1; public binds require
allowed_origins. Origin is validated structurally on every
method (POST/GET/DELETE/OPTIONS); literal Origin: null is
rejected unless explicitly allowed. CORS echoes the validated
origin (no wildcard); Access-Control-Allow-Headers is derived
from the configured auth provider.
	stdio transport (barrel_mcp_stdio).

Wire-level conformance
	POSTed JSON-RPC requests return either a JSON envelope or an SSE
stream. Notifications and POSTed responses to server-initiated
requests return HTTP 202 with empty body.
	Missing Mcp-Session-Id on a non-initialize request → 400;
unknown id → 404.
	MCP-Protocol-Version validated server-side: missing falls back
to the session-stored negotiated version; unsupported → 400.
	JSON-RPC id must be string or integer; null and other shapes
rejected with -32600. Top-level JSON arrays (batches) explicitly
rejected — MCP removed batching.
	notifications/cancelled aborts the in-flight tool call; the
cancelled HTTP request closes with 200 and an empty body (no
JSON-RPC envelope, per spec).
	Per-session SSE ring buffer (default 256 entries) for
Last-Event-ID replay; out-of-window ids surface a synthetic
notifications/replay_truncated event.

Registries
	Tools — handlers may be arity 1 or arity 2. Arity-2
handlers receive a Ctx map with session_id, request_id,
progress_token, the inbound meta map (the spec's _meta
extension hook), and an emit_progress function.
Meta-bearing return shapes ({result_meta, Result, Map},
{structured_meta, Data, Content, Map},
{tool_error, Content, Map}) attach _meta to the response
envelope.
	Resources — text/binary content, MIME types,
notifications/resources/updated for live updates. Handlers
may return a single block (#{text := _} /
#{blob := _, mimeType := _}, with optional mimeType and
annotations) or a list of pre-built content blocks for
multi-part responses.
	Resource templates — RFC 6570 URI templates, surfaced via
resources/templates/list. resources/read against a URI
matching a registered template auto-expands the variables
(Level 1, simple {var} substitutions) and routes to the
template handler with the substituted values in Args.
	Prompts — multi-message conversation templates with
arguments.
	Completions — keyed by {prompt, Name, Arg} or
{resource_template, Uri, Arg}; advertised via the
completions capability when at least one is registered.
	All registrations accept optional title and icons.
	Tool, resource, prompt, and resource-template registrations
also accept annotations — a free-form map surfaced verbatim
under annotations in the matching */list payload. Tools use
readOnlyHint, destructiveHint, idempotentHint,
openWorldHint; resources/prompts/templates use audience
(["user" | "assistant"]) and priority (0..1).

Tool features
	Return shapes: plain (text / map / list / image), {tool_error, Content} (→ isError: true), {structured, Data} /
{structured, Data, Content} (→ structuredContent).
	validate_input and validate_output opt-in schema validation
via barrel_mcp_schema.
	long_running => true returns a taskId immediately and runs
the worker in the background. Backed by barrel_mcp_tasks —
surfaces tasks/list, tasks/get, tasks/cancel, and
notifications/tasks/status.
	Cancellation: cooperative arity-2 handlers see
{cancel, RequestId} in their mailbox; arity-1 handlers run to
completion but their result is discarded.
	Progress: handlers call (maps:get(emit_progress, Ctx))(Done, Total, MessageOrUndef); out-of-band code can use
barrel_mcp:notify_progress/3,4.

Sessions
	ETS tables are protected; mutators run in
barrel_mcp_session's gen_server.
	Mcp-Session-Id lifecycle with TTL-based cleanup.
	Server-to-client sampling (sampling/createMessage),
elicitation (elicitation/create), roots query (roots/list),
and resource update notifications.

Authentication
	Providers: barrel_mcp_auth_bearer, barrel_mcp_auth_apikey,
barrel_mcp_auth_basic, barrel_mcp_auth_none,
barrel_mcp_auth_custom.
	Hashing: barrel_mcp_auth_basic:hash_password/1,2 defaults to
PBKDF2-SHA256 (100k iterations, 16-byte salt).
barrel_mcp_auth_apikey:hash_key/2 produces a peppered HMAC-SHA-256
digest. Both verifiers accept legacy hex SHA-256 digests for one
release. All comparisons are constant-time.
	OAuth 2.0 Protected Resource Metadata (RFC 9728): pass
resource_metadata => #{resource, authorization_servers} to
start_http_stream/1 / start_http/1 to expose
/.well-known/oauth-protected-resource and have the bearer
challenge emit WWW-Authenticate: Bearer ... resource_metadata="<URL>" so MCP clients auto-discover the
authorization server.

Server-to-client primitives
	Façade	Effect
	barrel_mcp:notify_resource_updated/1,2	notifications/resources/updated to every subscriber.
	barrel_mcp:notify_progress/3,4	notifications/progress to a session.
	barrel_mcp:notify_log/3,4	notifications/message (server log stream) to a session, filtered against the session's logging/setLevel.
	barrel_mcp:notify_list_changed/1	notifications/tools/list_changed, .../resources/list_changed, or .../prompts/list_changed to every active SSE session. Auto-emitted on reg_*/unreg_*.
	barrel_mcp:sampling_create_message/3	Server→client sampling/createMessage (requires the client to declare sampling capability).
	barrel_mcp:elicit_create/3	Server→client elicitation/create to ask the host for structured user input (requires the client to declare elicitation capability).
	barrel_mcp:roots_list/1,2	Server→client roots/list to enumerate the host's available roots (requires the client to declare roots capability).
	barrel_mcp_tasks:create/3, finish/3, fail/3, cancel/2	Long-running operation lifecycle.

Client (barrel_mcp_client)
barrel_mcp_client is a supervised gen_statem that holds one
connection to one MCP server and routes the protocol surface defined
by the spec.
Transports
	Transport	Module	Notes
	Streamable HTTP	barrel_mcp_client_http	POST with application/json, text/event-stream, SSE on POST and on a long-lived GET, Mcp-Session-Id capture, MCP-Protocol-Version after init, DELETE on close, 401 retry through barrel_mcp_client_auth.
	stdio	barrel_mcp_client_stdio	Subprocess line-delimited JSON-RPC.

Protocol coverage (Phase A — shipped)
	Targets 2025-11-25; negotiates downward through 2025-06-18,
2025-03-26, 2024-11-05.
	initialize with spec-shaped capability objects; notifications/initialized (the spec name).
	tools/list, tools/call, resources/list, resources/read,
resources/templates/list, resources/subscribe,
resources/unsubscribe, prompts/list, prompts/get,
completion/complete, logging/setLevel, ping,
tasks/list, tasks/get, tasks/cancel, tasks/result.
	Task statuses on the wire: working, completed, failed,
cancelled. Task timestamps (createdAt, lastUpdatedAt) are
RFC 3339 strings.
	Pagination via cursor / nextCursor. Single page by default
(want_cursor => true to follow paging by hand). The sugar helpers
list_tools_all/1, list_resources_all/1,
list_resource_templates_all/1, list_prompts_all/1,
tasks_list_all/1 walk every page via
barrel_mcp_pagination:walk/1.
	Cancellation: barrel_mcp_client:cancel/2 sends
notifications/cancelled and unblocks the caller.
	Roots changes: barrel_mcp_client:notify_roots_list_changed/1
emits notifications/roots/list_changed to inform the server
the host's roots have changed. The server may re-issue
roots/list.
	Progress: pass progress_token to call_tool/4 and the caller
receives {mcp_progress, Token, Params} for every matching
notifications/progress until the request settles.
	Periodic ping: opt-in via ping_interval (and
ping_failure_threshold) in the connect spec; the connection is
closed with reason ping_failed after the configured number of
consecutive failures.
	Server→client requests dispatched through the
barrel_mcp_client_handler behaviour. {reply, _, _},
{error, _, _, _}, and {async, Tag, _} reply forms; the host
later calls barrel_mcp_client:reply_async/3.
	Server→client notifications routed to handler;
notifications/resources/updated also forwarded to subscribers.

Federation
	barrel_mcp_clients registers one supervised client per
caller-chosen ServerId. Looked up via:	barrel_mcp:start_client/2
	barrel_mcp:stop_client/1
	barrel_mcp:whereis_client/1
	barrel_mcp:list_clients/0


	Tool-name namespacing across servers is host policy and is not
enforced by the library.

Auth
	barrel_mcp_client_auth behaviour.
	barrel_mcp_client_auth_bearer: static token.
	barrel_mcp_client_auth_oauth: OAuth 2.1 + PKCE.	Discovery: parse_www_authenticate/1, discover_protected_resource/1
(RFC 9728), discover_authorization_server/1 (RFC 8414 with OpenID
Connect fallback).
	PKCE: gen_code_verifier/0, code_challenge/1 (S256),
build_authorization_url/2.
	Token endpoint: exchange_code/2, refresh_token/2,
client_credentials/2, token_exchange/2, jwt_bearer/2.
All attach the RFC 8707 resource parameter; confidential
clients use HTTP Basic.
	Dynamic Client Registration (RFC 7591):
register_client/2 posts client metadata to the AS's
registration_endpoint and returns the AS's response
(client_id, optional client_secret, ...). Hosts then feed
the credentials into one of the connect-spec auth entries.
	Client Credentials grant (MCP ext-auth extension) for
unattended agent hosts: pass
auth => {oauth_client_credentials, Config} on the connect
spec. Required keys: token_endpoint, client_id. Authenticate
with client_secret (HTTP Basic) or client_assertion
(private_key_jwt, RFC 7523). Optional scopes, resource.
The library fetches the token eagerly during init/1 and
re-acquires via the same grant on 401 — no refresh_token
needed.
	Enterprise-Managed Authorization (MCP ext-auth EMA) for
SSO-driven hosts: pass auth => {oauth_enterprise, Config}.
Required keys: idp_token_endpoint, as_token_endpoint,
client_id, subject_token (an IdP ID Token / SAML
assertion the host obtained out of band),
subject_token_type, audience, resource. The handle
chains RFC 8693 token-exchange at the IdP through
RFC 7523 jwt-bearer at the AS and re-walks the same chain
on 401. An expired subject token surfaces as
{error, subject_token_expired} so the host can re-acquire
from the IdP.
	As an auth handle: when used through auth => {oauth, Config} on
the client spec, the library attaches Authorization: Bearer ...
on every request and runs the refresh-token grant on 401 if a
refresh_token was supplied. The interactive authorization-code
redirect stays a host concern.



Multi-server agent aggregator (barrel_mcp_agent)
Sits on top of barrel_mcp_clients and turns the federation
registry into a single namespaced tool catalog the host can hand
to an LLM, plus a router that dispatches a model's tool call back
to the right MCP server.
	list_tools/0,1 — aggregated tools/list across every
registered client; tool names rewritten to
<<"ServerId<sep>ToolName">> (default separator :).
	to_anthropic/0,1, to_openai/0,1 — the same catalog in the
matching provider shape.
	call_tool/2,3 — parses the namespaced name, routes to the
right client. Errors {error, no_separator | unknown_server}
cover the parse / lookup paths.

LLM provider bridge (barrel_mcp_tool_format)
Translates MCP tool maps to the shapes the major LLM provider
APIs expect, and translates a model's tool-call back into the
(Name, Arguments) pair barrel_mcp_client:call_tool/4 consumes.
	to_anthropic/1, to_openai/1 — MCP tool → provider tool.
	from_anthropic_call/1, from_openai_call/1 — provider call →
{Name, Args}. Accepts both parsed maps and JSON-string
arguments (the OpenAI wire shape).

Schema validation (barrel_mcp_schema)
Pure-Erlang JSON Schema subset validator hosts can use to pre-flight
LLM-generated tool args before calling the server. Covers type,
properties, required, enum, items, oneOf/anyOf/allOf,
additionalProperties: false, string minLength/maxLength/pattern,
number bounds, and array minItems/maxItems/uniqueItems.
case barrel_mcp_schema:validate(Args, ToolInputSchema) of
    ok -> barrel_mcp_client:call_tool(Pid, Name, Args);
    {error, Errors} -> reject(Errors)
end.
Notes on past roadmap items
	Periodic deadline timer. Earlier docs flagged a missing
global sweep for in-flight requests with infinity timeout.
In practice the client applies ?DEFAULT_REQUEST_TIMEOUT
(30s) per request unless the caller explicitly passes
timeout => infinity, so the default loop is already
time-bounded. Overriding an explicit infinity from a
background sweep would surprise callers who deliberately
disabled the deadline; the per-request timer is the right
hook.

	Client-side Last-Event-ID resume. Earlier docs said the
transport tracked the id but did not replay on reconnect.
This actually does work today within the transport process
lifetime: handle_sse_done schedules reopen_sse which
preserves sse_last_event_id, and start_get_sse re-adds
the last-event-id header on the new GET. A full client
restart (gen_statem crash) does lose the cursor, but that
flow re-initializes the session anyway, so a fresh stream
is the correct outcome.




  

    Streamable HTTP Transport

MCP Streamable HTTP transport for Claude Code (and any other MCP
client speaking the same protocol).
Overview
The Streamable HTTP transport implements MCP 2025-11-25 and
negotiates downward to 2025-06-18, 2025-03-26, and
2024-11-05 based on the client's initialize request. It is the
transport expected by Claude Code's --transport http option.
This transport supports:
	POST for client requests with JSON or SSE streaming responses.
	GET for server-to-client notification streams (SSE).
	DELETE for session termination.
	OPTIONS for CORS preflight.
	Session management via Mcp-Session-Id header.
	Replay on reconnect via Last-Event-ID.
	Origin validation with operator-controlled allow-list.

The built-in server is built on the h1 and h2 libraries, not
Cowboy. A cleartext bind speaks HTTP/1.1; a TLS bind serves both
HTTP/1.1 and HTTP/2 on the same port, chosen per connection by ALPN.
The protocol logic lives in a transport-neutral engine
(barrel_mcp_http_engine), so the same Streamable HTTP behaviour can
be driven by another HTTP stack (see
Embedding in another HTTP server).
Starting the Server
%% Basic start
barrel_mcp:start_http_stream(#{port => 9090}).

%% With API key authentication
barrel_mcp:start_http_stream(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{
            keys => #{
                <<"my-api-key">> => #{subject => <<"user1">>}
            }
        }
    }
}).

%% With session management disabled
barrel_mcp:start_http_stream(#{
    port => 9090,
    session_enabled => false
}).

%% With HTTPS/TLS
barrel_mcp:start_http_stream(#{
    port => 9443,
    ssl => #{
        certfile => "/path/to/cert.pem",
        keyfile => "/path/to/key.pem"
    }
}).
Options
	Option	Type	Default	Description
	port	pos_integer()	9090	Port number.
	ip	inet:ip_address()	{127,0,0,1}	Bind address. Default is loopback. Public binds require allowed_origins (see below).
	auth	map()	#{}	Authentication configuration.
	session_enabled	boolean()	true	Enable session management.
	ssl	map()	undefined	TLS configuration.
	allowed_origins	[binary()] | any	loopback set on loopback bind; required on public bind	List of allowed Origin values, structurally matched (scheme + host + port). Use any to disable validation. The literal <<"null">> may be included to allow sandboxed-frame origins.
	allow_missing_origin	boolean()	true on loopback, false otherwise	Whether to accept requests with no Origin header. Non-browser clients typically don't send one.
	sse_buffer_size	pos_integer()	256	Per-session ring buffer of recent SSE events for Last-Event-ID replay.

Security defaults
The transport binds to 127.0.0.1 by default. Public binds (any
non-loopback IP) require an explicit allowed_origins; the start
function refuses with {error, allowed_origins_required}
otherwise. This avoids accidental exposure to DNS-rebinding and
CORS-style attacks.
%% Public bind — must list allowed origins explicitly.
{ok, _} = barrel_mcp:start_http_stream(#{
    port => 9090,
    ip => {0, 0, 0, 0},
    allowed_origins => [<<"https://app.example.com">>]
}).
CORS responses echo the validated Origin (no wildcard) and add
Vary: Origin. The Access-Control-Allow-Headers list is
derived from the configured auth provider via the optional
auth_headers/1 callback on barrel_mcp_auth, so a custom
header_name on barrel_mcp_auth_apikey flows through both the
preflight allow-list and the request handler's header extraction.
Claude Code Integration
After starting the server, add it to Claude Code:
# Without authentication
claude mcp add my-server --transport http http://localhost:9090/mcp

# With API key authentication
claude mcp add my-server --transport http http://localhost:9090/mcp \
  --header "X-API-Key: my-api-key"

# With bearer token
claude mcp add my-server --transport http http://localhost:9090/mcp \
  --header "Authorization: Bearer my-token"

To verify the connection:
claude mcp list

Authentication
All authentication providers from barrel_mcp are supported:
No Authentication (Default)
barrel_mcp:start_http_stream(#{port => 9090}).
API Key
barrel_mcp:start_http_stream(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{
            keys => #{<<"key-123">> => #{subject => <<"user">>}}
        }
    }
}).
Bearer Token (JWT)
barrel_mcp:start_http_stream(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_bearer,
        provider_opts => #{
            secret => <<"your-jwt-secret">>
        }
    }
}).
Basic Auth
barrel_mcp:start_http_stream(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_basic,
        provider_opts => #{
            credentials => #{<<"admin">> => <<"password">>}
        }
    }
}).
Session Management
When session_enabled is true (default), the server tracks client sessions:
	Each client receives an Mcp-Session-Id header in responses
	Sessions expire after 30 minutes of inactivity (configurable via session_ttl env)
	GET requests open SSE streams for server notifications
	DELETE requests terminate sessions

Session Lifecycle
	First Request: Client sends POST without session ID
	Session Created: Server responds with Mcp-Session-Id: mcp_<hex>
	Subsequent Requests: Client includes Mcp-Session-Id header
	Termination: Client sends DELETE with session ID

Configuring Session TTL
%% In sys.config
{barrel_mcp, [
    {session_ttl, 3600000}  %% 1 hour in milliseconds
]}.
Server-Sent Events (SSE)
Clients that accept text/event-stream can receive streaming responses:
Request Format
POST /mcp HTTP/1.1
Accept: text/event-stream, application/json
Content-Type: application/json
Mcp-Session-Id: mcp_abc123

{"jsonrpc": "2.0", "method": "tools/list", "id": 1}
Response Format
HTTP/1.1 200 OK
Content-Type: text/event-stream
Mcp-Session-Id: mcp_abc123

id: 1706345678901234
data: {"jsonrpc": "2.0", "result": {...}, "id": 1}
HTTPS/TLS
For production deployments, enable HTTPS:
barrel_mcp:start_http_stream(#{
    port => 9443,
    ssl => #{
        certfile => "/path/to/fullchain.pem",
        keyfile => "/path/to/privkey.pem",
        cacertfile => "/path/to/chain.pem"  %% optional
    }
}).
A TLS bind advertises ALPN h2 and http/1.1, so the one port
serves HTTP/2 clients and HTTP/1.1 clients alike; certfile/keyfile
(and optional cacertfile) are file paths. Then use the HTTPS URL
with Claude Code:
claude mcp add my-server --transport http https://my-server.example.com:9443/mcp

CORS and request validation
The server validates Origin on every request method (POST, GET,
DELETE, OPTIONS) and replies with 403 on mismatch. When the
request's Origin validates, the response includes:
	Access-Control-Allow-Origin: <validated origin>
	Vary: Origin
	Access-Control-Allow-Methods: POST, GET, DELETE, OPTIONS
	Access-Control-Allow-Headers: content-type, accept, mcp-session-id, mcp-protocol-version, last-event-id plus any
auth headers declared by the configured provider.
	Access-Control-Expose-Headers: www-authenticate, mcp-session-id, mcp-protocol-version

When the request has no Origin header (typical of non-browser
clients) and allow_missing_origin is true, the response omits
Access-Control-Allow-Origin entirely rather than synthesising a
value.
Wire-level conformance
The transport implements MCP 2025-11-25 conformance points
explicitly:
	Wire	Behaviour
	MCP-Protocol-Version request header	Required after init. Unsupported value → 400; missing falls back to the session-stored negotiated version; pre-init missing assumes 2025-03-26.
	Mcp-Session-Id request header	Required on every non-initialize request when session_enabled is true. Missing → 400; unknown id → 404. initialize is the only request that creates a session.
	Notifications and posted server-bound responses	HTTP 202 Accepted, empty body.
	JSON-RPC ids	Must be string or integer. null or any other shape → -32600 Invalid Request.
	JSON-RPC batches	Top-level JSON arrays explicitly rejected with -32600 (MCP removed batching).
	notifications/cancelled	Cancels the in-flight tool call; the originating HTTP request closes with 200 and an empty body (no JSON-RPC envelope, per spec).

The session, subscription, and pending-request ETS tables are
protected; mutators run in the session manager so a non-owning
process cannot tamper with the table.
Protocol Differences
vs Legacy HTTP Transport
	Feature	Legacy (start_http)	Streamable (start_http_stream)
	Protocol Version	2024-11-05	2025-11-25 (negotiates downward)
	Claude Code	Not supported	Supported
	Sessions	No	Yes
	SSE Responses	No	Yes
	GET for streams	No	Yes
	DELETE for cleanup	No	Yes
	Origin validation	Yes	Yes

When to Use
	Use start_http_stream for Claude Code integration
	Use start_http for simple JSON-RPC clients
	Use start_stdio for Claude Desktop integration

Embedding in another HTTP server
The built-in h1/h2 server is one binding over a transport-neutral
engine, barrel_mcp_http_engine. If you already run an HTTP server
(for example the Livery web framework) you can serve MCP through it by
calling the engine directly, with no second listener.
For each request, read the method, path, headers and body, then call:
barrel_mcp_http_engine:handle(Method, Path, Headers, Body, Responder, Config).
	Method is the request method binary (<<"POST">>, <<"GET">> …).
	Path is the request target (a query string is allowed; the engine
strips it).
	Headers is a [{binary(), binary()}] proplist; lookups are
case-insensitive.
	Body is the full request body (<<>> when there is none).

Responder is a map of closures the engine uses to send the response,
so it never touches a socket:
#{reply        => fun(Status, Headers, Body) -> ok end,
  stream_start => fun(Status, Headers) -> ok end,
  stream_chunk => fun(Data) -> ok | {error, term()} end,
  stream_end   => fun() -> ok end}
Headers passed back to the closures is a [{binary(), binary()}]
proplist. A normal response is a single reply; a Server-Sent-Events
response is stream_start, then repeated stream_chunk, then
stream_end. handle/6 runs in the calling process; for a long-lived
GET SSE stream it blocks until the session ends or the host signals a
disconnect by sending the calling process the message mcp_disconnect.
Config is the engine configuration. Build it the way the bindings do
(barrel_mcp_http_stream:start/1 for mode => stream,
barrel_mcp_http:start/1 for mode => simple) using the shared
helpers:
Config = #{
    mode => stream,            %% or `simple'
    auth_config =>
        barrel_mcp_http_engine:init_auth(#{provider => barrel_mcp_auth_none}),
    session_enabled => true,
    allowed_origins => any,    %% or a resolved allow-list
    allow_missing_origin => true,
    sse_buffer_size => 256,
    resource_metadata => undefined
}.
The engine handles routing (/mcp, /, and
/.well-known/oauth-protected-resource), sessions, CORS, Origin
validation, authentication and async tool calls, identically to the
built-in server.
Example: Complete Server
-module(my_mcp_server).
-export([start/0]).

start() ->
    %% Start the application
    application:ensure_all_started(barrel_mcp),

    %% Register tools
    barrel_mcp:reg_tool(<<"greet">>, ?MODULE, greet, #{
        description => <<"Greet someone">>,
        input_schema => #{
            <<"type">> => <<"object">>,
            <<"properties">> => #{
                <<"name">> => #{<<"type">> => <<"string">>}
            }
        }
    }),

    %% Start streamable HTTP server
    {ok, _} = barrel_mcp:start_http_stream(#{
        port => 9090,
        auth => #{
            provider => barrel_mcp_auth_apikey,
            provider_opts => #{
                keys => #{<<"test-key">> => #{subject => <<"tester">>}}
            }
        }
    }),

    io:format("MCP server running on http://localhost:9090/mcp~n"),
    io:format("Add to Claude Code:~n"),
    io:format("  claude mcp add my-server --transport http http://localhost:9090/mcp --header \"X-API-Key: test-key\"~n").

greet(Args) ->
    Name = maps:get(<<"name">>, Args, <<"World">>),
    <<"Hello, ", Name/binary, "!">>.
See Also
	Authentication Guide
	Tools, Resources & Prompts
	MCP Specification



  

    stdio Transport

The stdio transport enables MCP communication over stdin/stdout, which is the
transport used by Claude Desktop and other MCP clients that spawn server processes.
Overview
Unlike HTTP transport, stdio transport:
	Uses newline-delimited JSON-RPC messages.
	Runs as a child process spawned by the MCP client.
	Is ideal for local integrations (no network overhead).
	Is the primary transport for Claude Desktop.

The same registries (tools, resources, prompts, resource
templates, completions, tasks) work over stdio. Tool handlers may
be arity 1 or arity 2 ((Args, Ctx)) — the emit_progress and
cancellation hooks in Ctx interleave on stdout in stdio just
like they do on the SSE channel for HTTP. See the
Tools guide for the handler shape.
Quick Start
1. Create an Escript
#!/usr/bin/env escript
%%! -pa _build/default/lib/*/ebin

-module(my_mcp_server).
-mode(compile).

main(_Args) ->
    %% Start the application
    application:ensure_all_started(barrel_mcp),
    barrel_mcp_registry:wait_for_ready(),

    %% Register your tools
    barrel_mcp:reg_tool(<<"hello">>, my_mcp_server, hello, #{
        description => <<"Say hello">>,
        input_schema => #{
            <<"type">> => <<"object">>,
            <<"properties">> => #{
                <<"name">> => #{
                    <<"type">> => <<"string">>,
                    <<"description">> => <<"Name to greet">>
                }
            }
        }
    }),

    %% Start stdio server (blocks until stdin closes)
    barrel_mcp:start_stdio().

hello(Args) ->
    Name = maps:get(<<"name">>, Args, <<"World">>),
    <<"Hello, ", Name/binary, "!">>.
2. Make it Executable
chmod +x my_mcp_server

3. Configure Claude Desktop
Edit your claude_desktop_config.json:
macOS: ~/Library/Application Support/Claude/claude_desktop_config.json
Windows: %APPDATA%\Claude\claude_desktop_config.json
Linux: ~/.config/claude/claude_desktop_config.json
{
  "mcpServers": {
    "my-erlang-server": {
      "command": "/absolute/path/to/my_mcp_server",
      "args": []
    }
  }
}
4. Restart Claude Desktop
After saving the config, restart Claude Desktop. Your MCP server will be available.
Blocking vs Supervised Mode
Blocking Mode
Use barrel_mcp:start_stdio/0 when you want the server to run in the current process:
main(_Args) ->
    setup_tools(),
    barrel_mcp:start_stdio().  %% Blocks here
This is ideal for escripts and simple applications.
Supervised Mode
Use barrel_mcp:start_stdio_link/0 when you want the server supervised:
-module(my_app_sup).
-behaviour(supervisor).
-export([init/1]).

init([]) ->
    %% Ensure tools are registered first
    setup_tools(),

    Children = [
        #{id => mcp_stdio,
          start => {barrel_mcp, start_stdio_link, []},
          restart => permanent,
          type => worker}
    ],
    {ok, {#{strategy => one_for_one}, Children}}.
Protocol Details
Message Format
Each message is a single line of JSON (newline-delimited):
{"jsonrpc":"2.0","id":1,"method":"initialize","params":{"protocolVersion":"2024-11-05"}}\n
Supported Methods
The stdio transport supports all MCP methods:
	initialize / initialized - Connection lifecycle
	tools/list / tools/call - Tool operations
	resources/list / resources/read - Resource operations
	prompts/list / prompts/get - Prompt operations
	ping - Keep-alive

Notifications
MCP notifications (methods without id) don't receive responses:
{"jsonrpc":"2.0","method":"notifications/initialized"}\n
Building Releases
For production use, build an Erlang release instead of an escript.
Using rebar3 Release
	Add to rebar.config:

{relx, [
    {release, {my_mcp_server, "1.0.0"}, [my_app, barrel_mcp]},
    {mode, prod},
    {extended_start_script, true}
]}.
	Create your main module:

-module(my_mcp_main).
-export([start/0]).

start() ->
    %% Called when release starts
    setup_tools(),
    barrel_mcp:start_stdio().
	Configure your app to call this on start:

%% In your application module
start(_Type, _Args) ->
    %% Start your supervisor
    {ok, Sup} = my_app_sup:start_link(),

    %% If running in MCP mode, start stdio
    case application:get_env(my_app, mcp_mode, false) of
        true -> spawn(fun my_mcp_main:start/0);
        false -> ok
    end,

    {ok, Sup}.
	Build and run:

rebar3 release
_build/default/rel/my_mcp_server/bin/my_mcp_server foreground

Debugging
Testing Locally
You can test your stdio server manually:
# Start your server
./my_mcp_server

# Then type JSON-RPC messages (each on one line):
{"jsonrpc":"2.0","id":1,"method":"initialize","params":{}}
{"jsonrpc":"2.0","id":2,"method":"tools/list"}
{"jsonrpc":"2.0","id":3,"method":"tools/call","params":{"name":"hello","arguments":{"name":"Erlang"}}}

Logging
Since stdout is used for MCP responses, use stderr for debugging:
debug(Msg) ->
    io:format(standard_error, "[DEBUG] ~s~n", [Msg]).
Or use Erlang's logger to a file:
%% Configure in your app startup
logger:add_handler(file_handler, logger_std_h, #{
    config => #{file => "/tmp/mcp_server.log"}
}).
Common Issues
Server not appearing in Claude Desktop:
	Check config file path and JSON syntax
	Use absolute path to executable
	Restart Claude Desktop after config changes

"Command not found" errors:
	Ensure the executable has the shebang line
	Check file permissions (chmod +x)
	Use absolute paths in config

No responses:
	Ensure all tools are registered before start_stdio/0
	Check stderr for errors

Environment Variables
Claude Desktop passes environment variables to your server:
%% Access them in your code
HomeDir = os:getenv("HOME"),
PathVar = os:getenv("PATH").
You can also configure environment in claude_desktop_config.json:
{
  "mcpServers": {
    "my-server": {
      "command": "/path/to/my_mcp_server",
      "args": [],
      "env": {
        "MY_CONFIG": "/path/to/config.json",
        "DEBUG": "true"
      }
    }
  }
}
Working Directory
The working directory is typically the user's home directory or where
Claude Desktop was launched. To ensure consistent behavior:
%% Set a known working directory
file:set_cwd("/path/to/my/app"),

%% Or use absolute paths for all file operations
ConfigPath = filename:join([os:getenv("HOME"), ".config", "myapp"]).
See Also
	Getting Started - Basic setup
	Tools, Resources & Prompts - MCP primitives
	barrel_mcp_stdio module documentation



  

    Authentication

barrel_mcp provides a pluggable authentication system following OAuth 2.1 patterns
as recommended by the MCP specification. Authentication is optional and configurable
per HTTP server instance.
Overview
Authentication in barrel_mcp is handled by providers - modules implementing the
barrel_mcp_auth behaviour. The library includes several built-in providers:
	Provider	Use Case
	barrel_mcp_auth_none	No authentication (default)
	barrel_mcp_auth_bearer	JWT tokens or opaque Bearer tokens
	barrel_mcp_auth_apikey	API key authentication
	barrel_mcp_auth_basic	HTTP Basic authentication

Bearer Token Authentication
The most common pattern for MCP servers, supporting both JWT and opaque tokens.
JWT with HS256
barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_bearer,
        provider_opts => #{
            %% HMAC secret for HS256 signature verification
            secret => <<"your-256-bit-secret-key-here">>,

            %% Optional: Validate issuer claim
            issuer => <<"https://auth.example.com">>,

            %% Optional: Validate audience claim
            audience => <<"https://api.example.com">>,

            %% Optional: Clock skew tolerance in seconds (default: 60)
            clock_skew => 120,

            %% Optional: Custom scope claim name (default: <<"scope">>)
            scope_claim => <<"permissions">>
        },
        %% Optional: Require specific scopes
        required_scopes => [<<"mcp:read">>, <<"mcp:write">>]
    }
}).
JWT with RS256/ES256 (Custom Verifier)
For asymmetric algorithms, provide a custom verifier function:
%% Using jose library for RS256
Verifier = fun(Token) ->
    try
        JWK = jose_jwk:from_pem_file("public_key.pem"),
        case jose_jwt:verify(JWK, Token) of
            {true, {jose_jwt, Claims}, _} -> {ok, Claims};
            {false, _, _} -> {error, invalid_token}
        end
    catch
        _:_ -> {error, invalid_token}
    end
end,

barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_bearer,
        provider_opts => #{verifier => Verifier}
    }
}).
Opaque Tokens (Token Introspection)
For tokens that require server-side validation:
Verifier = fun(Token) ->
    %% Call your auth server's introspection endpoint
    case httpc:request(post, {
        "https://auth.example.com/introspect",
        [{"Authorization", "Bearer " ++ SecretKey}],
        "application/x-www-form-urlencoded",
        "token=" ++ binary_to_list(Token)
    }, [], []) of
        {ok, {{_, 200, _}, _, Body}} ->
            Claims = jsx:decode(list_to_binary(Body), [return_maps]),
            case maps:get(<<"active">>, Claims, false) of
                true -> {ok, Claims};
                false -> {error, invalid_token}
            end;
        _ ->
            {error, invalid_token}
    end
end,

barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_bearer,
        provider_opts => #{verifier => Verifier}
    }
}).
API Key Authentication
Simple and effective for server-to-server communication.
Static Key Map
barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{
            keys => #{
                <<"ak_prod_abc123">> => #{
                    subject => <<"service-a">>,
                    scopes => [<<"read">>, <<"write">>],
                    metadata => #{team => <<"platform">>}
                },
                <<"ak_prod_xyz789">> => #{
                    subject => <<"service-b">>,
                    scopes => [<<"read">>]
                }
            }
        }
    }
}).
Hashed Keys (Recommended for Production)
The recommended format is a peppered HMAC-SHA-256 digest. The
pepper is a server-side secret mixed into the hash so a leak of
the stored hash table on its own isn't enough to forge keys.
Pepper = <<"random-32-byte-pepper-loaded-from-env">>,
Hash1 = barrel_mcp_auth_apikey:hash_key(<<"ak_prod_abc123">>,
                                         #{pepper => Pepper}),
Hash2 = barrel_mcp_auth_apikey:hash_key(<<"ak_prod_xyz789">>,
                                         #{pepper => Pepper}),

barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{
            keys => #{
                Hash1 => #{subject => <<"service-a">>},
                Hash2 => #{subject => <<"service-b">>}
            },
            hash_keys => true,
            pepper => Pepper
        }
    }
}).
Stored values look like
<<"hmac-sha256$<base64-encoded-hash>">>.
For verification outside the auth pipeline (config tooling,
tests), use barrel_mcp_auth_apikey:verify_key/2 — it does a
constant-time comparison and accepts both the new format and
legacy unsalted hex SHA-256 digests for one release.
Migrating from legacy SHA-256
hash_key/1 (no pepper) still produces the legacy hex SHA-256
format and is still accepted by the verifier. To migrate:
	Generate a pepper and load it from a secret store.
	Re-hash each existing key with hash_key/2.
	Replace the entries in your keys map.
	Drop the legacy entries.

Custom Header Name
barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{
            header_name => <<"X-Service-Key">>,  %% Custom header
            keys => #{<<"my-key">> => #{subject => <<"service">>}}
        }
    }
}).
Dynamic Key Validation
Verifier = fun(ApiKey) ->
    case my_db:lookup_api_key(ApiKey) of
        {ok, #{user_id := UserId, scopes := Scopes}} ->
            {ok, #{subject => UserId, scopes => Scopes}};
        not_found ->
            {error, invalid_credentials}
    end
end,

barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_apikey,
        provider_opts => #{verifier => Verifier}
    }
}).
Basic Authentication
HTTP Basic auth - simple but requires TLS in production.
Static Credentials
barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_basic,
        provider_opts => #{
            credentials => #{
                <<"admin">> => <<"secret123">>,
                <<"readonly">> => <<"viewer456">>
            },
            realm => <<"MCP Server">>
        }
    }
}).
Hashed Passwords
hash_password/1,2 defaults to PBKDF2-SHA256 (100k
iterations, random 16-byte salt). Stored values look like
<<"pbkdf2-sha256$<iters>$<base64(salt)>$<base64(hash)>">>.
%% Hash passwords once, store the resulting binary, use that in
%% the credentials map.
AdminHash = barrel_mcp_auth_basic:hash_password(<<"secret123">>),
UserHash  = barrel_mcp_auth_basic:hash_password(<<"viewer456">>),

barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_basic,
        provider_opts => #{
            credentials => #{
                <<"admin">>    => AdminHash,
                <<"readonly">> => UserHash
            },
            hash_passwords => true
        }
    }
}).
hash_password/2 accepts an options map:
	algorithm — pbkdf2-sha256 (default) or sha256-hex (legacy,
for migration only).
	iterations — PBKDF2 iteration count (default 100000).

The verification path is the public verify_password/2. It
accepts the modern format and legacy hex SHA-256 digests for one
release; the legacy code path logs a deprecation warning on every
match.
With Scopes and Metadata
barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_basic,
        provider_opts => #{
            credentials => #{
                <<"admin">> => #{
                    password => <<"secret123">>,
                    scopes => [<<"read">>, <<"write">>, <<"admin">>],
                    metadata => #{role => <<"administrator">>}
                }
            }
        }
    }
}).
Custom Authentication Provider
Implement the barrel_mcp_auth behaviour for custom authentication:
-module(my_oauth_provider).
-behaviour(barrel_mcp_auth).

-export([init/1, authenticate/2, challenge/2]).

%% Initialize provider state
init(Opts) ->
    ClientId = maps:get(client_id, Opts),
    ClientSecret = maps:get(client_secret, Opts),
    IntrospectUrl = maps:get(introspect_url, Opts),
    {ok, #{
        client_id => ClientId,
        client_secret => ClientSecret,
        introspect_url => IntrospectUrl
    }}.

%% Authenticate a request
authenticate(Request, State) ->
    Headers = maps:get(headers, Request, #{}),
    case barrel_mcp_auth:extract_bearer_token(Headers) of
        {ok, Token} ->
            introspect_token(Token, State);
        {error, no_token} ->
            {error, unauthorized}
    end.

%% Generate challenge response for failed auth
challenge(unauthorized, State) ->
    Realm = maps:get(realm, State, <<"mcp">>),
    {401, #{
        <<"www-authenticate">> => <<"Bearer realm=\"", Realm/binary, "\"">>
    }, <<"{\"error\":\"unauthorized\"}">>};
challenge(invalid_token, _State) ->
    {401, #{
        <<"www-authenticate">> => <<"Bearer error=\"invalid_token\"">>
    }, <<"{\"error\":\"invalid_token\"}">>};
challenge(insufficient_scope, _State) ->
    {403, #{
        <<"www-authenticate">> => <<"Bearer error=\"insufficient_scope\"">>
    }, <<"{\"error\":\"insufficient_scope\"}">>}.

%% Internal: Token introspection
introspect_token(Token, #{introspect_url := Url} = State) ->
    %% Your introspection logic here
    case call_introspection_endpoint(Token, Url, State) of
        {ok, #{<<"active">> := true} = Claims} ->
            {ok, #{
                subject => maps:get(<<"sub">>, Claims),
                scopes => parse_scopes(maps:get(<<"scope">>, Claims, <<>>)),
                claims => Claims
            }};
        _ ->
            {error, invalid_token}
    end.
Accessing Auth Info in Handlers
After successful authentication, auth info is available in the request:
my_tool_handler(Args) ->
    case maps:get(<<"_auth">>, Args, undefined) of
        undefined ->
            %% No auth (using barrel_mcp_auth_none)
            do_something_anonymous();
        #{subject := Subject, scopes := Scopes} = AuthInfo ->
            %% Authenticated request
            case lists:member(<<"admin">>, Scopes) of
                true -> do_admin_action(Subject);
                false -> do_user_action(Subject)
            end
    end.
Error Responses
Authentication failures return proper HTTP status codes and WWW-Authenticate headers:
	Error	Status	Description
	unauthorized	401	No credentials provided
	invalid_token	401	Token is malformed or signature invalid
	expired_token	401	Token has expired
	invalid_credentials	401	Wrong username/password or API key
	insufficient_scope	403	Token lacks required scopes

OAuth grant flows
barrel_mcp_client ships three grants plus a registration
pre-step. Pick by who is in the loop and when:
Authorization Code + PKCE — interactive
For flows where a real user authorises the host. Browser
redirect, PKCE prevents code interception, refresh on 401.
User    Host                   AS                  MCP server
 │  click "connect"             │                       │
 │──►│ redirect + PKCE          │                       │
 │   │─────────────────────────►│                       │
 │   │ login + consent          │                       │
 │   │◄─────────────────────────│                       │
 │   │ exchange_code +          │                       │
 │   │ code_verifier            │                       │
 │   │─────────────────────────►│                       │
 │   │ access_token +           │                       │
 │   │ refresh_token            │                       │
 │   │◄─────────────────────────│                       │
 │   │ Bearer access_token                              │
 │   │──────────────────────────────────────────────────►│
 │   │       (on 401: refresh_token grant)              │
auth => {oauth, #{
    access_token   => <<"...">>,            % required
    refresh_token  => <<"...">>,            % optional, enables refresh
    token_endpoint => <<"https://idp/oauth/token">>,
    client_id      => <<"...">>,
    client_secret  => <<"...">>,            % optional confidential client
    resource       => <<"https://mcp/...">>,
    scopes         => [<<"mcp.read">>, <<"mcp.write">>]
}}
The host drives the browser dance and feeds the resulting tokens
in. The library handles the refresh.
Client Credentials — unattended (M2M)
For agent hosts running without a human. The host already has
its own credentials.
Host                              AS                  MCP server
 │  POST /token                    │                       │
 │  grant=client_credentials       │                       │
 │  + Basic <client_id:secret>     │                       │
 │────────────────────────────────►│                       │
 │  access_token                   │                       │
 │◄────────────────────────────────│                       │
 │  Bearer access_token                                    │
 │────────────────────────────────────────────────────────►│
 │             (on 401: re-acquire via same grant)         │
auth => {oauth_client_credentials, #{
    token_endpoint   => <<"https://idp/oauth/token">>,
    client_id        => <<"my-agent-host">>,
    client_secret    => <<"...">>,          % OR client_assertion (JWT)
    resource         => <<"https://mcp/...">>,
    scopes           => [<<"mcp.read">>]
}}
Eager fetch on init — a misconfigured client fails up front.
Re-acquires via the same grant on 401. No refresh_token is
involved.
Enterprise-Managed Authorization — SSO chain
For SSO-driven hosts. The user already has a session at the org
IdP; their identity flows into a short-lived MCP access token
without re-prompting. Two-step chain (RFC 8693 → RFC 7523).
User    Host        IdP             AS                MCP server
 │  active SSO session              │                       │
 │  ID Token                        │                       │
 │ ─────────►│                      │                       │
 │           │ POST /token          │                       │
 │           │ grant=token-exchange │                       │
 │           │ subject_token=<id_token>                     │
 │           │ audience=<AS issuer> │                       │
 │           │ resource=<MCP url>   │                       │
 │           │─────────────────────►│                       │
 │           │ ID-JAG (signed JWT)  │                       │
 │           │◄─────────────────────│                       │
 │           │ POST /token          │                       │
 │           │ grant=jwt-bearer     │                       │
 │           │ assertion=<ID-JAG>   │                       │
 │           │─────────────────────►│                       │
 │           │  access_token        │                       │
 │           │◄─────────────────────│                       │
 │           │  Bearer access_token                         │
 │           │─────────────────────────────────────────────►│
 │           │ (on 401: re-walk the chain)                 │
 │           │ (id_token expires →                         │
 │           │  {error, subject_token_expired})            │
auth => {oauth_enterprise, #{
    idp_token_endpoint => <<"https://idp/oauth/token">>,
    as_token_endpoint  => <<"https://as/oauth/token">>,
    client_id          => <<"...">>,
    client_secret      => <<"...">>,        % OR client_assertion
    subject_token      => <IdToken>,        % from IdP, opaque
    subject_token_type =>
        <<"urn:ietf:params:oauth:token-type:id_token">>, % or saml2
    audience           => <<"https://as">>,
    resource           => <<"https://mcp/...">>,
    scopes             => [<<"mcp.read">>]
}}
The library treats subject_token as opaque — both OIDC and
SAML modes hit the same code path. The browser flow at the IdP
stays a host concern.
Dynamic Client Registration — pre-step
Not a grant. Used before any of the others when the host
doesn't have a client_id yet (fresh deployment, distributed
host, dev sandbox). RFC 7591.
Host                                    AS
 │  POST /register                       │
 │  { client_name, redirect_uris,        │
 │    grant_types, ... }                 │
 │──────────────────────────────────────►│
 │  { client_id, client_secret?,         │
 │    client_id_issued_at, ... }         │
 │◄──────────────────────────────────────│
{ok, Info} = barrel_mcp_client_auth_oauth:register_client(
    <<"https://idp/oauth/register">>,
    #{<<"client_name">> => <<"my-host">>, ...}),
ClientId     = maps:get(<<"client_id">>, Info),
ClientSecret = maps:get(<<"client_secret">>, Info, undefined).
For protected registration endpoints (RFC 7591 section 3) pass
the AS-issued initial access token via register_client/3:
{ok, Info} = barrel_mcp_client_auth_oauth:register_client(
    <<"https://idp/oauth/register">>,
    #{<<"client_name">> => <<"my-host">>, ...},
    #{initial_access_token => <<"...">>}).
Feed the returned credentials into one of the grants above. The
library does not persist them; that's a host concern.
Where they overlap on the wire
All grants hit the same OAuth-server token endpoint with
application/x-www-form-urlencoded bodies. Confidential clients
authenticate with HTTP Basic; private_key_jwt clients pass a
client_assertion instead. RFC 8707 resource is attached on
every grant. The MCP 2025-11-25 auth sub-spec layers
RFC 9728 PRM on
top so any of the grants can be auto-discovered from a 401
response.
When to pick which
	Situation	Use
	Real user, browser available, host wants their identity	auth_code ({oauth, ...})
	Background agent / cron / unattended host	client_credentials ({oauth_client_credentials, ...})
	Enterprise SSO; user identity must flow to MCP	enterprise_managed ({oauth_enterprise, ...})
	No client_id yet	register_client/2 first, then one of the above

OAuth Protected Resource Metadata (RFC 9728)
For OAuth-protected deployments, MCP clients auto-discover the
authorization server by:
	Hitting any endpoint and receiving a 401 with
WWW-Authenticate: Bearer ... resource_metadata="<URL>".
	Following the URL to fetch the
RFC 9728
Protected Resource Metadata document.
	Reading authorization_servers from the document and
completing the OAuth dance against one of them.

The server side of this loop is opt-in via a resource_metadata
option on barrel_mcp:start_http_stream/1 (and
start_http/1):
{ok, _} = barrel_mcp:start_http_stream(#{
    port => 8080,
    auth => #{provider => barrel_mcp_auth_bearer,
              provider_opts => #{secret => Secret}},
    resource_metadata => #{
        resource              => <<"http://localhost:8080/mcp">>,
        authorization_servers => [<<"https://idp.example.com">>]
    }
}).
When set:
	/.well-known/oauth-protected-resource is served by the HTTP
transport as a JSON metadata document.
	The bearer challenge on 401 emits
resource_metadata="<absolute PRM URL>". The PRM URL is
derived from resource by default; pass
metadata_url => <<"https://...">> in the option map to
override.

The audience-claim string in state.resource (used for token
verification by barrel_mcp_auth_bearer) is unaffected; only
the wire emission of WWW-Authenticate changed.
The client side is implemented by
barrel_mcp_client_auth_oauth:parse_www_authenticate/1 and
discover_protected_resource/1 — together with the server side
above, the MCP authorization sub-spec discovery flow works
end-to-end.
Dynamic Client Registration (RFC 7591)
Hosts that don't have a pre-issued client_id for the target
authorization server can register one programmatically. The AS
metadata document advertises the endpoint via
registration_endpoint.
{ok, Info} = barrel_mcp_client_auth_oauth:register_client(
    <<"https://idp.example.com/oauth/register">>,
    #{<<"client_name">>   => <<"my-mcp-host">>,
      <<"redirect_uris">> => [<<"http://localhost:5173/callback">>],
      <<"grant_types">>   => [<<"authorization_code">>,
                              <<"refresh_token">>],
      <<"response_types">> => [<<"code">>],
      <<"token_endpoint_auth_method">> => <<"none">>}).

%% Info now contains:
%%   #{<<"client_id">>           => <<"...">>,
%%     <<"client_secret">>       => <<"...">>,   % if confidential
%%     <<"client_id_issued_at">> => 1700000000,  % UNIX epoch
%%     ...}
Feed the returned credentials into a subsequent
{oauth, ...} / {oauth_client_credentials, ...} connect spec.
This stays a standalone exchanger — the library doesn't persist
the issued credentials. That's a host concern (file, DB, secret
manager).
Security Best Practices
	Always use TLS in production
	Hash stored credentials using the provided hash functions
	Use short-lived tokens with refresh capability
	Validate audience claims to prevent token misuse
	Implement rate limiting at the transport layer
	Log authentication failures for security monitoring



  

    Custom Authentication

barrel_mcp provides a simplified way to integrate custom authentication systems via barrel_mcp_auth_custom. This is useful when you have an existing key management system and want to use it for MCP HTTP authentication.
Simple Interface
Instead of implementing the full barrel_mcp_auth behaviour (with init/1, authenticate/2, and challenge/2), you only need two functions:
-module(my_auth).
-export([init/1, authenticate/2]).

%% Initialize authentication state
-spec init(Opts :: map()) -> {ok, State :: term()}.
init(_Opts) ->
    {ok, #{}}.

%% Authenticate a token
-spec authenticate(Token :: binary(), State :: term()) ->
    {ok, AuthInfo :: map(), NewState :: term()} |
    {error, Reason :: term(), NewState :: term()}.
authenticate(Token, State) ->
    case my_key_store:validate(Token) of
        {ok, KeyInfo} ->
            AuthInfo = #{
                subject => maps:get(user_id, KeyInfo),
                scopes => maps:get(permissions, KeyInfo, [])
            },
            {ok, AuthInfo, State};
        error ->
            {error, invalid_token, State}
    end.
Usage
Configure the HTTP server to use your custom auth module:
barrel_mcp:start_http(#{
    port => 9090,
    auth => #{
        provider => barrel_mcp_auth_custom,
        provider_opts => #{
            module => my_auth,
            module_opts => #{} % passed to my_auth:init/1
        }
    }
}).
Token Extraction
barrel_mcp_auth_custom automatically extracts tokens from HTTP requests. It checks (in order):
	Authorization: Bearer <token> header
	X-API-Key: <token> header

The extracted token is passed to your authenticate/2 function.
AuthInfo Format
Your authenticate/2 function should return a map with:
	Key	Type	Description
	subject	binary	User/client identifier
	scopes	[binary]	List of permission scopes

Additional keys are preserved in the claims field.
Example: barrel_memory Integration
Here's how barrel_memory uses custom auth with its existing key system:
-module(barrel_memory_mcp_auth).
-export([init/1, authenticate/2]).

init(Opts) ->
    {ok, Opts}.

authenticate(Token, State) ->
    case barrel_memory_api_keys:validate_key(Token) of
        {ok, KeyInfo} ->
            AuthInfo = #{
                subject => maps:get(team_id, KeyInfo, <<"unknown">>),
                scopes => maps:get(permissions, KeyInfo, [])
            },
            {ok, AuthInfo, State};
        {error, invalid_key} ->
            {error, invalid_token, State}
    end.
Configuration in barrel_memory:
barrel_mcp:start_http(#{
    port => 9091,
    auth => #{
        provider => barrel_mcp_auth_custom,
        provider_opts => #{
            module => barrel_memory_mcp_auth
        }
    }
}).
Claude Code Configuration
Once your MCP server is running with custom auth, add it to Claude Code:
# With Bearer token
claude mcp add my-server --transport http http://localhost:9090/mcp \
  --header "Authorization: Bearer your-api-key"

# Or with X-API-Key header
claude mcp add my-server --transport http http://localhost:9090/mcp \
  --header "X-API-Key: your-api-key"

Error Handling
When authentication fails, barrel_mcp_auth_custom returns a 401 response with:
WWW-Authenticate: Bearer realm="mcp"
Your authenticate/2 can return any error reason - it will be logged but not exposed to clients.

